
Introduction
The BASIC assembler, as standard, does not have any support for true floating point instructions. You have the ability to convert integers to your implementation-defined 'floating point' and perform basic mathematics with them (most usually fixed point), but you cannot interact with a floating point co-processor and do things the 'native' way.
There are, however, patches which extend the things that the assembler can do - which include FP instructions. 
Parts of this documentation has been taken from the ARM Assembler manual. 
Please note that this describes the floating point implementation used with RISC OS and not the VFP provided with the more recent ARM processors (ARMv5 etc). 
  
How it works with the ARM
The ARM processor can interface with up to sixteen co-processors. The ARM3 and later have virtual co-processors within the ARM to handle internal control functions. But the first co-processor that was available was the floating point processor. This chip handles floating point maths to the IEEE standard. 
A standard ARM floating point instruction set has been defined, so that the code may be used across all RISC OS machines. If the actual hardware does not exist, then the instructions are trapped and executed by the floating point emulator module (FPEmulator). The program does not need to know whether or not the FP co-processor is present. The only real difference will be speed of execution. 
If you are interested in the co-processor aspect, read the document on co-processor access. 
Note: If a real FPU is attached, this will pick up on the unrecognised instructions and do the work itself. However in the case of systems such as the ARM7500FE, the work of the floating point unit is shared between hardware (for instructions like MUF (multiply)) and software (for instructions like LGN (log. to base e)). 
  
FP registers
The ARM IEEE FP system has eight high precision FP registers (F0 to F7). The register format is irrelevant as you cannot access those registers directly, the register is only 'visible' when it is transferred to memory or to an ARM register. In memory, an FP register consumes three words, but as the FP system will be reloading its own register, the format of these three words is considered irrelevant. 
There is also an FPSR (floating point status register) which, similar to the ARM's own PSR, holds the status information that an application might require. Each of the flags available has a 'trap' which allows the application to enable or disable traps associated with the given error. 
The FPSR also allows you to tell between different implementations of the FP system.
There may also be an FPCR (floating point control register). This holds information that the application should not access, such as flags to turn the FP unit on and off. Typically, hardware will have an FPCR, software will not. Do not attempt to use the FPCR - some parts of it are read-sensitive, so your reading it will affect the rest of the system (like FPE!). 
FP units can be software implementations such as the FPEmulator modules, hardware implementations such as the FP chip (and support code), or a combination of both.
The "most original" example of a 'both' that I can think of is the Warm Silence Software patch that will utilise the 80x87 chip on suitably equipped PC co-processor cards as a floating point processor for ARM FP operations. Talk about resource sharing...! 
The results are calculated as though it were infinite precision, then they are rounded to the length required. The rounding may be to nearest, to +infinity(P), to -infinity(M), or to zero. The default is rounding to nearest. If a tie, it will round to nearest even.
The working precision is 80 bits, comprising of a 64 bit mantissa, a 15 bit exponent, and a sign bit. Specific instructions that work with single precision may provide better performance in some implementations - notably fully-software-based ones. 
The FPSR contains the necessary status for the FP system. The IEEE flags are always present, but the result flags are only available after an FP compare operation. 
Floating point instructions should not be used from SVC mode. 
The FPSR is laid out as follows: 
31           24 23           16 15             8 7              0
System ID      | Trap enable   | System control | Exception flags
The defined system IDs are: 
&00   Old FPE  - FPE module prior to v4.00
&80   FPPC     - Interface between ARM and WE32206 (AT&T MAU)
&01   FPE 400  - FPE module v4.00 or later
&81   FPA      - ARM FPU
The Trap enable byte is: 
23 22 21   20     19     18     17     16
Reserved   INX    UFL    OFL    DVZ    IVO
If an exception flag bit is set following an operation, and the corresponding trap enable bit is set, then the exception trap will be taken. 
The System control byte is: 
15 14 13   12     11     10     9      8
Reserved   AC     EP     SO     NE     ND
(these bits have no meaning on Old FPE and FPPC systems) 
AC Use Alternative definition for C flag on compares 
When set, the ARM 'C' flag is set after a FP comparison to mean Greater Than or Equal or Unordered. When clear, 'C' is set for Greater Than or Equal. 
 
EP Use Expanded Packed decimal format 
When set, choosing Packed Decimal format will result in Expanded Packed Decimal to be used.
The PRM says "Use of this expanded format allows conversion from extended precision to packed decimal and back again without loss of accuracy." Given that Double Extended Precision uses three words, while Extended Packed Decimal uses four, why not use Double Extended if you need to keep precision?
I'm sure there's a reason for EPD - I just don't get it... 
 
SO Select Synchronous Operation of the FPU 
When set, all FP operations are executed synchronously and the ARM will be forced to busy-wait until the FPU is done.
There is a trade off. With synchronous operation, exceptions will be raised at the expected time and place, and the addresses of such exceptions will be correct. However, with async. operation, the ARM can actually get on and do stuff while the FPU does it's own stuff (provided, of course, the ARM isn't waiting on an FP result).
Obviously, this bit has no meaning on the software FP systems! 
 
NE NaN Exception 
The PRM says: "If this bit is set, then an attempt to store a signalling NaN that involves a change of format will cause an exception (for full IEEE compatibility).
I have no idea what that actually means (to me, a 'signalling NaN' brings to mind an old lady with big glasses waving a red lantern out the back of a steam train...).
The alternative? If this bit is clear, then no exception will occur - for compatibility with 'old FPE'. 
 
ND No Denormalised numbers 
The PRM says "If this bit is set, then the software will force all denormalised numbers to zero to prevent lengthy execution times when dealing with denormalised numbers."
Unfortunately, "the software" is ambiguous - I assume this means the FPE module does this. Would a hardware FPU do this as well?
I looked in my dictionary for a definition of 'denormalised' - it went from 'denominator' to 'denotation'. Oh well... 
Exception flags, the lower byte of the FPSR: 
7  6   5   4      3      2      1      0
Reserved   INX    UFL    OFL    DVZ    IVO
Whenever an exception condition arises, the appropriate cumulative exception flag in bits 0 to 4 will be set to 1. If the relevant trap enable bit is set, then an exception is also delivered to the user's program in a manner specific to the operating system. (Note that in the case of underflow, the state of the trap enable bit determines under which conditions the underflow flag will be set.) These flags can only be cleared by a WFS instruction. 
IVO InValid Operation 
The IVO flag is set when an operand is invalid for the operation to be performed. Invalid operations are: 
· Any operation on a trapping NaN (not-a-number) 
· Magnitude subtraction of infinities, eg + + - 
· Multiplication of 0 by 0 
· Division of 0/0 or x/0 
· x REM y where x = 8 or y = 0
(REM is the `remainder after floating point division' operator.) 
· Square root of any number < 0 (but (-0) = -0) 
· Conversion to integer or decimal when overflow, or a NaN operand make it impossible
If overflow makes a conversion to integer impossible, then the largest positive or negative integer is produced (depending on the sign of the operand) and IVO is signalled 
· Comparison with exceptions of Unordered operands 
· ACS, ASN, SIN, COS, TAN, LOG, LGN, POW, or RPW with invalid/incorrect arguments. 

 
DVZ DiVision by Zero 
The DVZ flag is set if the divisor is zero and the dividend a finite, non-zero number. A correctly signed infinity is returned if the trap is disabled. The flag is also set for LOG(0) and for LGN(0). Negative infinity is returned if the trap is disabled. 
 
OFL OverFLow 
The OFL flag is set whenever the destination format's largest number is exceeded in magnitude by the rounded result would have been were the exponent range unbounded. As overflow is detected after rounding a result, whether overflow occurs or not after some operations depends the rounding mode.
If the trap is disabled either a correctly signed infinity is returned, or the format's largest finite number. This depends on the rounding mode and floating point system used. 
 
UFL UnderFLow
Two correlated events contribute to underflow: 
· Tininess - the creation of a tiny non-zero result smaller in magnitude than the format's smallest normalised number. 
· Loss of accuracy - a loss of accuracy due to denormalisation that may be greater than would be caused by rounding alone. 
The UFL flag is set in different ways depending on the value of the UFL trap enable bit. If the trap is enabled, then the UFL flag is set when tininess is detected regardless of loss of accuracy. If the trap is disabled, then the UFL flag is set when both tininess and loss of accuracy are detected (in which case the INX flag is also set); otherwise a correctly signed zero is returned.
As underflow is detected after rounding a result, whether underflow occurs or not after some operations depends on the rounding mode. 
 
INX INeXact 
The INX flag is set if the rounded result of an operation is not exact (different from the value computable with infinite precision), or overflow has occurred while the OFL trap was disabled, or underflow has occurred while the UFL trap was disabled. OFL or UFL traps take precedence over INX.
The INX flag is also set when computing SIN or COS, with the exceptions of SIN(0) and COS(1).
The old FPE and the FPPC system may differ in their handling of the INX flag. Because of this inconsistency, it is recommended that you do not enable the INX trap. 
  
Precision and rounding
Precision is: 
· S  - single 
· D  - double 
· E  - double extended 
· P  - packed decimal 
· EP - extended packed decimal (if enabled) 
A precision must be specified. 
  
Rounding modes are: 
·    - nearest (no letter required) 
· P  - plus infinity 
· M  - minus infinity 
· Z  - zero 
If no rounding mode is specified, 'nearest' will be assumed. 
  
Data format
We shall briefly describe the Single and Double precision numbers as they appear in memory. If you require details of the other three formats, you are referred to appropriate documentation, such a the RISC OS 3 Programmer's Reference Manuals volume 4, pages 167 to 169. 
Because the floating point system is a little complex, we shall quickly look at how floating point typically operates.
As we all know, a computer can use bit patterns to represent numbers. Older machines could easily handle between 0 and 255, or 0 and 65535. The ARM processor can easily handle between zero and 4294967295. A 64 bit processor can easily handle between 0 and 18446744073709551615.
Obviously, by using other techniques, any system can store all sorts of numbers - BBC BASIC running on a 6502 didn't crash if you told it to count to 257, for example.
As the data widths get larger, and the numbers that can be handled in one go get larger, this still does not help with the simplest case of PI...
But... there's a solution. An eight-bit processor can handle numbers larger than 255 by using the simple formula ( (high_byte x 256) + low_byte ). Okay, things are more complex, but you get the idea.
So, if we want to store PI (we'll take PI as being 3.14159265), then why don't we simply store the number 314159265, and alongside it we'll store also a value saying that the 'real' decimal point should shift eight places to the left.
PI = 314159265.0 [<-8] = 31415926.5 [<-7] = ... = 3.14159265
In the above example, I've shown it after the first shift of the decimal point, just to give you a visual idea of how it works.
Unfortunately, the IEEE system isn't anywhere near this easy to understand. In fact, I've written it up below and given some examples, but I still don't get it... 
  
IEEE Single Precision
 31   30      23 22                  0
Sign | Exponent | (msb) Fraction (lsb)
· If the exponent and fraction are both 0, the number is ±0. 
· If the exponent is 0 and the fraction is non-zero, the number is ±0.fraction x 2-126. 
· If the exponent is 1-254, the number is ±1.fraction x 2exponent-127. 
· If the exponent is 255 and the fraction is 0, the number is ∞ (infinity). 
· If the exponent is 255 and the fraction is non-zero, a NaN is represented. If the msb of the fraction is set, it is a non-trapping NaN; else it is a trapping NaN. 
IEEE Double Precision
              31   30      20 19                  0
First word:  Sign | Exponent | (msb) Fraction (lsb)
Second word: (msb)          Fraction          (lsb)
· If the exponent and fraction are both 0, the number is ±0. 
· If the exponent is 0 and the fraction is non-zero, the number is ±0.fraction x 2-1022. 
· If the exponent is 1-2046, the number is ±1.fraction x 2exponent-1023. 
· If the exponent is 2047 and the fraction is 0, the number is ∞ (infinity). 
· If the exponent is 2047 and the fraction is non-zero, a NaN is represented. If the msb of the fraction is set, it is a non-trapping NaN; else it is a trapping NaN. 
To a non-mathematician (such as myself!), the system does not appear to make an awful lot of sense, for example: 
DIM code% 64
FOR opt% = 0 TO 2 STEP 2
  P% = code%
  [  OPT  opt%
     ext  1
     MVFS F0, #0.5      ; or whatever other value...
     STFS F0, store
     MOV  PC, R14
   .store
     DCD  0   ; only one word - we're using single precision
  ]
NEXT
CALL code%
Examining the memory location gives: 
   0     &00000000   exponent = 0    fraction = 0        sign = 0
   0.5   &40000000   exponent = 9    fraction = 0        sign = 0
   1     &3F800000   exponent = 254  fraction = 0        sign = 0
   2     &08B4000D   exponent = 34   fraction = 13       sign = 0
   5     &40A00000   exponent = 2    fraction = 2097152  sign = 1
   10    &41200000   exponent = 4    fraction = 2097152  sign = 1
Continuing, working up some code with the 'printlib' example supplied with the C/assembler development software, we arrive at: 
 -0.123456 is -1.2346E-1
   9.99996 is 1.0000E1
-0.0999998 is -1.0000E-1
  0.999997 is 1.0000E0
      -0.0 is 0
9.99999E99 is 1.0000E100
This apprarently does make sense - only not to me - so I cannot explain it! :-) 
  
The FP instruction set
[bookmark: ldf]
LDF<condition><precision><fp register>, <address>
Load Floating Point value
The address can be in the forms: 
· [Rn] 
· [Rn], #offset 
· [Rn, #offset] 
· [Rn, #offset]! 
This call is similar to LDR.
Your assembler may allow literals to be used, such as LDFS F0, [float_value] 
  
[bookmark: stf]
STF<condition><precision><fp register>, <address>
Store floating point value. The address can be in the forms: 
· [Rn] 
· [Rn], #offset 
· [Rn, #offset] 
· [Rn, #offset]! 
This call is similar to STR.
Your assembler may allow literals to be used, such as STFED F0, [float_value] 
  
[bookmark: lfm][bookmark: sfm]
LFM and SFM
These are similar in idea to LDM and STM, but they will not be described because some versions of FPEmulator do not support them. The FP module in RISC OS 3.1x (2.87) does, as do (I think) later versions. If you know they your software will only operate on a system that supports SFM, then use it. Otherwise you'll need to 'fake' it with a sequence of STFs. Likewise for LFM/LDF. 
  
[bookmark: flt]
FLT<condition><precision><rounding> <fp register>, <register>
FLT<condition><precision><rounding> <fp register>, #<value>
Convert integer to floating point, either an ARM register or an absolute value. 
  
[bookmark: fix]
FIX<condition><rounding> <register>, <fp register>
Convert floating point to integer. 
  
[bookmark: wfs]
WFS<condition> <register>
Write floating point status register with the contents of the ARM register specified. 
  
[bookmark: rfs]
RFS<condition> <register>
Read floating point status register into the ARM register specified. 
  
[bookmark: wfc]
WFC<condition> <register>
Write floating point control register with the contents of the ARM register specified.
Supervisor mode only, and only on hardware that supports it. 
  
[bookmark: rfc]
RFC<condition> <register>
Read floating point control register into the ARM register specified.
Supervisor mode only, and only on hardware that supports it. 
  
Floating point co-processor data operations:
The formats of these instructions are:
· binary_operation condition precision rounding Fdest, Fsource, Fsource 
· binary_operation condition precision rounding Fdest, Fdest, #value 
· unary_operation condition precision rounding Fdest, Fsource 
· unary_operation condition precision rounding Fdest, #value 
The #value constants should be 0, 1, 2, 3, 4, 5, 10, or 0.5. 
[bookmark: binop]
The binary operations are...
ADF - Add
DVF - Divide
FDV - Fast Divide - only defined to work with single precision
FML - Fast Multiply - only defined to work with single precision
FRD - Fast Reverse Divide - only defined to work with single precision
MUF - Multiply
POL - Polar Angle
POW - Power
RDF - Reverse Divide
RMF - Remainder
RPW - Reverse Power
RSF - Reverse Subtract
SUF - Subtract
[bookmark: unop]
The unary operations are...
ABS - Absolute Value
ACS - Arc Cosine
ASN - Arc Sine
ATN - Arc Tangent
COS - Cosine
EXP - Exponent
LOG - Logarithm to base 10
LGN - Logarithm to base e
MVF - Move
MNF - Move Negated
NRM - Normalise
RND - Round to integral value
SIN - Sine
SQT - Square Root
TAN - Tangent
URD - Unnormalised Round
[bookmark: cmf]
CMF<condition><precision><rounding> <fp register 1>, <fp register 2>
Compare FP register 2 with FP register 1.
The varient CMFE compares with exception. 
[bookmark: cnf]
CNF<condition><precision><rounding> <fp register 1>, <fp register 2>
Compare FP register 2 with the negative of FP register 1.
The varient CMFE compares with exception. 
Compares are provided with and without the exception that could arise if the numbers are unordered (ie one or both of them is not-a-number). To comply with IEEE 754, the CMF instruction should be used to test for equality (ie when a BEQ or BNE is used afterwards) or to test for unorderedness (in the V flag). The CMFE instruction should be used for all other tests (BGT, BGE, BLT, BLE afterwards). 
  
When the AC bit in the FPSR is clear, the ARM flags N, Z, C, V refer to the following after compares:
N = Less than
Z = Equal
C = Greater than, or equal
V = Unordered 
When the AC bit in the FPSR is clear, the ARM flags N, Z, C, V refer to the following after compares:
N = Less than
Z = Equal
C = Greater than, or equal
V = Unordered 
And when the AC bit is set, the flags refer to:
N = Less than
Z = Equal
C = Greater than, or equal, or unordered
V = Unordered 
  
In APCS code with objasm, to store a floating point value, you would use the directive DCF. You append 'S' for single precision, and 'D' for double. 
  
Examples
Here is a brief example. We MUL two numbers, but use the floating point unit instead of the ARM's multiplication instruction. This could be modified to multiply two floating point numbers, and give a floating point response, but as it is only a short example, it will simply use two integers. 
REM >fpmul
REM
REM Short example to multiply two integers via the
REM floating point unit. Totally pointless, but...

DIM code% 20

FOR loop% = 0 TO 2 STEP 2
  P% = code%
  [  OPT loop%

   .multiply
     FLTS   F0, R0
     FLTS   F1, R1
     FMLS   F2, F0, F1
     FIXS   R0, F2

     MOVS   PC, R14
  ]
NEXT

INPUT "First number  : "one%
INPUT "Second number : "two%

A% = one%
B% = two%
result% = USR(multiply)

PRINT "The result is "+STR$(result%)
END
There is no option to download this program, as standard BASIC won't touch it. However, you can include FP statements if you can 'build' the instructions.
Alternatively, you could use ExtBASasm by Darren Salt. 
This version will work in BASIC: 
REM >fpmul
REM
REM Short example to multiply two integers via the
REM floating point unit. Totally pointless, but...

DIM code% 20

FOR loop% = 0 TO 2 STEP 2
  P% = code%
  [  OPT loop%

   .multiply
     EQUD   &EE000110   ; FLTS F0, R2
     EQUD   &EE011110   ; FLTS F1, R1
     EQUD   &EE902101   ; FMLS F2, F0, F1
     EQUD   &EE100112   ; FIXS R0, F2

     MOVS   PC, R14
  ]
NEXT

INPUT "First number  : "one%
INPUT "Second number : "two%

A% = one%
B% = two%
result% = USR(multiply)

PRINT "The result is "+STR$(result%)
END
Download this example
  
Remember to use the appropriate precision for what you are doing. 
REM >precision
REM
REM Short example to show how data can be 'lost' due
REM to using incorrect precision.

ON ERROR PRINT REPORT$ + " at " + STR$(ERL/10) : END

DIM code% 64

FOR loop% = 0 TO 2 STEP 2
  P% = code%
  [  OPT loop%

     EXT 1

   .single_precision
     FLTS   F0, R0
     FIX    R0, F0
     MOV    PC, R14

   .double_precision
     FLTD   F0, R0
     FIX    R0, F0
     MOV    PC, R14

   .doubleext_precision
     FLTE   F0, R0
     FIX    R0, F0
     MOV    PC, R14
  ]
NEXT

A% = &1ffffff

PRINT "Original input is " + STR$~A%
PRINT "Single precision  " + STR$~(USR(single_precision))
PRINT "Double precision  " + STR$~(USR(double_precision))
PRINT "Double extended   " + STR$~(USR(doubleext_precision))
PRINT
END
The result of this program is: 
Original input is 1FFFFFF
Single precision  2000000
Double precision  1FFFFFF
Double extended   1FFFFFF
You don't need to use double precision everywhere, though, as it will be that much slower. Simply keep this in mind if you are dealing with large numbers. 
  
In order to test the actual speed differences, I wrote a test program: 
DIM code% 64

FOR loop% = 0 TO 2 STEP 2
  P% = code%
  [  OPT loop%

     MOV    R0, #23
     MOV    R1, #1<<16

   .timetest
     FLTD   F0, R0
     FLTD   F1, R0
     MUFD   F2, F0, F1
     SUBS   R1, R1, #1
     BNE    timetest

     MOV    PC, R14
  ]
NEXT

t% = TIME
CALL code%
PRINT "That took "+STR$(TIME - t%)+" centiseconds."
END
I tried various precisions, and also the fast multiply. It showed something interesting. So I tried multiplication, and addition. All with the same data (input 23). 
  
Here are my results for a million (roughly) convert-and-process operations. I've just timed my RiscPC and the times were MUCH slower - so I'm not entirely sure which system the timings below relate to - it did say "ARM710 processor, FPEmulator 4.14" but I doubt that... 
   Operation        Fast single   Single        Double        Double extended

   Multiplication   1731cs        1755cs        1965cs        1712cs

   Division         2169cs        2169cs        2618cs        2479cs

   Addition         n/a           1684cs        1899cs        1646cs
This seems to show that double extended precision is the fastest on my machine for a selection of operations. Thus, it is incorrect to simply assume more complexity takes longer time. My personal suspicion here is the internal format is double extended, thus working directly with it entails no loss due to converting the value to a different precision. 
Why do I doubt the above experiments? Simple. Here are the results for an ARM710 RiscPC using FPEmulator 4.14 (1.07Mz): 
   Operation        Fast single   Single        Double        Double extended

   Multiplication   112cs         112cs         110cs         111cs

   Division         138cs         139cs         153cs         159cs

   Addition         n/a           108cs         107cs         106cs
These results seem more consistent, so... :-) 
The moral here? Don't be afraid to experiment... 

[bookmark: adc]ADC : Addition with Carry
  ADC<suffix>  <dest>, <op 1>, <op 2>

                dest = op_1 + op_2 + carry
ADC will add the two operands, placing the result in the destination register. It uses a carry bit, so can add numbers larger than 32 bits. The following example will add two 128 bit numbers.
128 bit result: Registers 0, 1, 2, and 3
128 bit first: Registers 4, 5, 6, and 7
128 bit second: Registers 8, 9, 10, and 11. 
  ADDS    R0, R4, R8              ; Add low words
  ADCS    R1, R5, R9              ; Add next word, with carry
  ADCS    R2, R6, R10             ; Add third word, with carry
  ADCS    R3, R7, R11             ; Add high word, with carry
If doing addition such as this, don't forget to set the S suffix so that the status of the Carry flag is updated. 
  
  
[bookmark: add]ADD : Addition
  ADD<suffix>  <dest>, <op 1>, <op 2>

                dest = op_1 + op_2
ADD will add the two operands, placing the result in the destination register. Operand 1 is a register, operand 2 can be a register, shifted register, or an immediate value: 
  ADD     R0, R1, R2              ; R0 = R1 + R2
  ADD     R0, R1, #256            ; R0 = R1 + 256
  ADD     R0, R2, R3,LSL#1        ; R0 = R2 + (R3 << 1)
The addition may be performed on signed or unsigned numbers. 
  
  
[bookmark: and]AND : Logical AND
  AND<suffix>  <dest>, <op 1>, <op 2>

                dest = op_1 AND op_2
AND will perform a logical AND between the two operands, placing the result in the destination register; this is useful for masking the bits you wish to work on. Operand 1 is a register, operand 2 can be a register, shifted register, or an immediate value: 
  AND     R0, R0, #3              ; R0 = Keep bits zero and one of R0,
                                         discard the rest.
An AND table (result = both): 
  Op_1   Op_2   Result

  0      0      0
  0      1      0
  1      0      0
  1      1      1
  
  
[bookmark: bic]BIC : Bit Clear
  BIC<suffix>  <dest>, <op 1>, <op 2>

                dest = op_1 AND (!op_2)
BIC is a way to clear bits within a word, a sort of reverse OR.
Operand two is a 32 bit bit mask. If a bit is set in the mask, it will be cleared. Unset mask bits indicate bits to be left alone. 
  BIC     R0, R0, #%1011          ; Clear bits zero, one, and
                                    three in R0. Leave the
                                    remaining bits alone.
A BIC table: 
  Op_1   Op_2   Result

  0      0      0
  0      1      0
  1      0      1
  1      1      0
  
  
[bookmark: eor]EOR : Logical Exclusive OR
  EOR<suffix>  <dest>, <op 1>, <op 2>

                dest = op_1 EOR op_2
EOR will perform a logical Exclusive OR between the two operands, placing the result in the destination register; this is useful for inverting certain bits. Operand 1 is a register, operand 2 can be a register, shifted register, or an immediate value: 
  EOR     R0, R0, #3              ; Invert bits zero and one in R0
An EOR table (result = either, but not both): 
  Op_1   Op_2   Result

  0      0      0
  0      1      1
  1      0      1
  1      1      0
  
  
[bookmark: mov]MOV : Move
  MOV<suffix>  <dest>, <op 1>

                dest = op_1
MOV loads a value into the destination register, from another register, a shifted register, or an immediate value.
You can specify the same register for the effect of a NOP instruction, or you can shift the same register if you choose: 
  MOV     R0, R0                  ; R0 = R0... NOP instruction

  MOV     R0, R0, LSL#3           ; R0 = R0 * 8
If R15 is the destination, the program counter or flags can be modified. This is used to return to calling code, by moving the contents of the link register into R15: 
  MOV     PC, R14                 ; Exit to caller

  MOVS    PC, R14                 ; Exit to caller preserving flags
                                    (not 32-bit compliant)
  
  
[bookmark: mvn]MVN : Move Negative
  MVN<suffix>  <dest>, <op 1>

                dest = !op_1
MVN loads a value into the destination register, from another register, a shifted register, or an immediate value.
The difference is the bits are inverted prior to moving, thus you can move a negative value into a register.
Due to the way this works (two's complement), you want to move one less than the required number: 
  MVN     R0, #4                  ; R0 = -5

  MVN     R0, #0                  ; R0 = -1
  
  
[bookmark: orr]ORR : Logical OR
  ORR<suffix>  <dest>, <op 1>, <op 2>

                dest = op_1 OR op_2
OR will perform a logical OR between the two operands, placing the result in the destination register; this is useful for setting certain bits to be set. Operand 1 is a register, operand 2 can be a register, shifted register, or an immediate value: 
  ORR     R0, R0, #3              ; Set bits zero and one in R0
An OR table (result = either or both): 
  Op_1   Op_2   Result

  0      0      0
  0      1      1
  1      0      1
  1      1      1
  
  
[bookmark: rsb]RSB : Reverse Subtraction
  RSB<suffix>  <dest>, <op 1>, <op 2>

                dest = op_2 - op_1
SUB will subtract operand one from operand two, placing the result in the destination register. Operand 1 is a register, operand 2 can be a register, shifted register, or an immediate value: 
  RSB     R0, R1, R2              ; R0 = R2 - R1
  RSB     R0, R1, #256            ; R0 = 256 - R1
  RSB     R0, R2, R3,LSL#1        ; R0 = (R3 << 1) - R2
The subtraction may be performed on signed or unsigned numbers. 
  
  
[bookmark: rsc]RSC : Reverse Subtraction with Carry
  RSC<suffix>  <dest>, <op 1>, <op 2>

                dest = op_2 - op_1 - !carry
This is the same as SBC, except the operands are subtracted the other way around. 
  
  
[bookmark: sbc]SBC : Subtraction with Carry
  SBC<suffix>  <dest>, <op 1>, <op 2>

                dest = op_1 - op_2 - !carry
SBC will subtract the two operands, placing the result in the destination register. It uses the carry bit to represent 'borrow', so can subtract numbers larger than 32bits.
SUB and SBC generate the Carry flag the wrong way around, if a borrow is required then the carry flag is UNSET. Thus, this instruction requires a NOT Carry flag - it inverts the flag automatically during the instruction. 
  
  
[bookmark: sub]SUB : Subtraction
  SUB<suffix>  <dest>, <op 1>, <op 2>

                dest = op_1 - op_2
SUB will subtract operand two from operand one, placing the result in the destination register. Operand 1 is a register, operand 2 can be a register, shifted register, or an immediate value: 
  SUB     R0, R1, R2              ; R0 = R1 - R2
  SUB     R0, R1, #256            ; R0 = R1 - 256
  SUB     R0, R2, R3,LSL#1        ; R0 = R2 - (R3 << 1)
The subtraction may be performed on signed or unsigned numbers. 
  
  
[bookmark: swp]SWP : Swap
  SWP<suffix>  <dest>, <op 1>, [<op 2>]
SWP will: 
· Load a word from memory, address pointed to by operand two, and put that word in the destination register. 
· Store the contents of register operand one to that same address. 
If the destination and operand one are the same register, then the contents of the register and the contents of the memory location given will be swapped.
If the B suffix is set, then a byte will be transferred, otherwise a word will be transferred.

The BASIC assembler and many DDE assemblers provide you with a selection of pseudo-instructions. These are instructions which are not actually understood by the processor, but are converted into something that may be understood, or otherwise used to make the code 'correct'. They exist to make your programming simpler. 
  
[bookmark: adr]ADR : load ADRess
   ADR<suffix> <register>, <label>
This loads the address referred into the given register: 
   00008FE4                    OPT     l%
   00008FE4 E28F0004           ADR     R0, text
   00008FE8 EF000002           SWI     "OS_Write0"
   00008FEC E1A0F00E           MOV     PC, R14
   00008FF0                    .text
   00008FF0                    EQUS    "Hello!" + CHR$13 + CHR$10 + CHR$0
   00008FFC                    ALIGN
The following code has the exact same effect: 
   00008FE4                    OPT     l%
   00008FE4 E28F0004           ADD     R0, R15, #4
   00008FE8 EF000002           SWI     "OS_Write0"
   00008FEC E1A0F00E           MOV     PC, R14
   00008FF0                    .text
   00008FF0                    EQUS    "Hello!" + CHR$13 + CHR$10 + CHR$0
   00008FFC                    ALIGN
Indeed, a disassembly of either of the above would show: 
   *MemoryI 8FE4 +18
   00008FE4 :  E28F0004 : ..�â : ADR     R0,&00008FF0
   00008FE8 :  EF000002 : ...ï : SWI     "OS_Write0"
   00008FEC :  E1A0F00E : .ð á : MOV     PC,R14
   00008FF0 :  6C6C6548 : Hell : STCVSTL CP5,C6,[R12],#-&120 ; =288
   00008FF4 :  0A0D216F : o!.. : BEQ     &003515B8
   00008FF8 :  00000000 : .... : DCD     &00000000
  
ADR is a useful instruction, though, as you do not need to worry about R15's offset (ie, why do we only add four?) nor do you need to calculate the offset over a block of code. Simply use ADR Rx, label and the assembler will sort it out for you using ADD, SUB, MOV or MVN as is best appropriate.
The limiting factor, however, is you can only reference within a 4096 byte range (not strictly true, it typically uses ADD or SUB with eight bits rotated by a multiple of two, but for argument's sake we'll assume the range is 4K). 
Therefore... 
  
  
[bookmark: adrl]ADRL : load ADRess Long
   ADRL<suffix> <register>, <label>
This is not supported by BASIC's assembler, though some extensions add it. 
The ADRL instruction uses a combination of ADR and ADD, or ADR and SUB, to generate a wider range of addresses which may be reached. However this always uses two instructions, so a more workable arrangement may be to try to restructure your code so that a normal ADR will work. 
There is also, in some assemblers, an ADRX which uses three instructions to address even more locations. 
Here is a macro for ADRL that you may use to address within a 64K range: 
   DEFFNadrl(register%, location%, pass%)
   REM Uses two instructions to load an address within a 64K range
   LOCAL progcnt%
   progcnt% = P%
   IF (location% > progcnt%) THEN
     [ OPT     pass% ; positive offset
       ADR     register%, (location% - ((location% - progcnt%) AND &FFFFFF00))
       ADD     register%, register%, #((location% - progcnt%) AND &FFFFFF00)
     ]
   ELSE
     [ OPT     pass%
       ADR     register%, (location% + ((progcnt% - location%) AND &FFFFFF00))
       SUB     register%, register%, #((progcnt% - location%) AND &FFFFFF00)
     ]
   ENDIF
   = 0
  
  
[bookmark: align]ALIGN : ALIGN pointers
   ALIGN
The ALIGN instruction sets P% (and, if required, O%) to be aligned on a word boundary. This is usually required following a string or one or more bytes of data, and should be used before further code is assembled. 
The BASIC assembler is quite smart, and experience shows that it will deal with alignment issues for you, if you neglect to do it... 
   00008FF4                    OPT     l%
   00008FF4 E28F0004           ADR     R0, text
   00008FF8 EF000002           SWI     "OS_Write0"
   00008FFC EA000004           B       carryon
   00009000                    .text
   00009000                    EQUS    "unaligned text!!!" + CHR$0
   00009012                    .carryon
   00009014 E1A0F00E           MOV     PC, R14
  
  
[bookmark: dcx]DCx : initialise data storage
   DCx <value>
There is no DCx instruction; the small 'x' denotes a range of possibilities. These are: 
   DCB    Reserve a byte (8 bit value)
   DCW    Reserve a half-word (16 bit value)
   DCD    Reserve a word (32 bit value)
   DCS    Reserve up to 255 bytes as required by given string

   DCF    Reserve space for a floating point value (not in 'normal' BASIC assembler).
For example: 
   .start_counter
     DCB     1

   .pointer
     DCD     0

   .error_block
     DCD     17
     DCS     "Uh-oh! It all went wrong!" + CHR$0
     ALIGN
  
  
[bookmark: equx]EQUx : initialise data storage
   EQUx <value>
objasm does not provide the EQUx versions, but you can add them as macros if you do not like DCx. 
There is no EQUx instruction; the small 'x' denotes a range of possibilities. These are: 
   EQUB   Reserve a byte (8 bit value)
   EQUW   Reserve a half-word (16 bit value)
   EQUD   Reserve a word (32 bit value)
   EQUS   Reserve up to 255 bytes as required by given string

   DCF    Reserve space for a floating point value (not in 'normal' BASIC assembler).
This is exactly the same as DCx (above), simply understood by a different name.
You can use '=' as a short-hand version of EQUB. 
  
  
NOP : No OPeration
   NOP
As you'll see, sometimes it is necessary to do nothing for a cycle after a mode change or the like. This is where NOP comes in, so there is a use for an instruction that does nothing! 
Actually, this instruction is translated to MOV R0, R0 which does something, but it is pretty pointless and has no side effects; thus it is an admirable NOP. 
Please note that some (dated) sources may suggest you use an instruction with an NV condition code, like MOVNV R0, R0. Not only is this pointless (why bother not executing it if executing it doesn't have any side effects?), but also the 'NV' condition code was been withdrawn... well, for at least a decade. With all the Thumb extensions, I would not be surprised if the bit pattern used by 'NV' meant something entirely different on newer ARM processors. 
  
OPT : set assembler OPTions
   OPT <value>
This sets various assembler options; produce listing? offset assembly? ignore errors? and the like. You will find more detail in the document devoted to OPT. 
Not discussed here, but OPT also has meanings to objasm. Refer to your assembler programming user guide (probably on the C/C++ DDE CD-ROM as a PDF file) for further details. 

The ARM processor incorporates a barrel shifter that can be used with the data processing instructions (ADC, ADD, AND, BIC, CMN, CMP, EOR, MOV, MVN, ORR, RSB, SBC, SUB, TEQ, TST). You can also use the barrel shifter to affect the index value in LDR/STR operations. 
There are six mnemonics for the different shift types: 
  LSL  Logical Shift Left
  ASL  Arithmetic Shift Left
  LSR  Logical Shift Right
  ASR  Arithmetic Shift Right
  ROR  Rotate Right
  RRX  Rotate Right with Extend
ASL and LSL are the same, and may be freely interchanged. 
You can specify a shift with an immediate value (from zero to thirty one), or by a register which contains a value between zero and thirty one. 
  
[bookmark: lsl][bookmark: asl]Logical or Arithmetic Shift Left
  Rx, LSL #n    or
  Rx, ASL #n    or
  Rx, LSL Rn    or
  Rx, ASL Rn
The contents of Rx will be taken and shifted to a more significant position by the amount specified by 'n' or in the register Rn. The least significant bits are filled with zeroes. The high bits shifted off to the left are discarded, except for the notional bit thirty three (i.e., the last bit to be shifted off) which becomes the value of the carry flag on exit from the barrel shifter. 
Consider the following: 
  MOV    R1, #12
  MOV    R0, R1, LSL#2
On exit, R0 is 48. The instructions forming the sum R0 = #12, LSL#2 is equivalent to the BASIC R0 = 12 << 2 
  
[bookmark: lsr]Logical Shift Right
  Rx, LSR #n    or
  Rx, LSR Rn
This does the notional opposite of shifting left. Everything is shifted to the right, to a less significant position. It is the same as register = value >>> shift. 
  
[bookmark: asr]Arithmetic Shift Right
  Rx, ASR #n    or
  Rx, ASR Rn
This is similar to LSR, with the exception that the high bits are filled with the value of bit 31 of the register being shifted (Rx), in order to preserve the sign in 2's complement maths. It is the same as register = value >> shift. 
  
[bookmark: ror]Rotate Right
  Rx, ROR #n    or
  Rx, ROR Rn
Rotate Right is similar to a Logical Shift Right, except the bits which would normally be shifted off the right are replaced on the left, this the bits 'rotate'. An operation where the amount to be shifted by is 32 would result in the output being identical to the input as all bits would have been shifted by 32 places, i.e., right back to where they started! 
  
[bookmark: rrx]Rotate Right with Extend
  Rx, RRX
This is a ROR#0 operation, which rotates one place to the right - the difference is that the processor's Carry flag is used to provide a 33 bit quantity to be shifted.
Thanks to Ian Jeffray for the correction. 

[bookmark: b]B : Branch
  B<suffix>    <address>
B is the simplest branch. Upon encountering a B instruction, the ARM processor will jump immediately to the address given, and resume execution from there.
Note that the actual value stored in the branch instruction is an offset from the current value of R15; rather than an absolute address. 
On other processors, you might often find code such as: 
  OPT 1
  LDA &70
  CMP #0
  BEQ Zero
  STA &72
 .Zero RTS
(from the Acorn Electron User Guide issue 1 page 213) 
On the ARM processor, that would become something like: 
  OPT     1
  ADR     R1, #&70
  LDR     R0, [R1]
  CMP     #0
  BEQ     Zero
  STR     R0, [R1, #2]
 .Zero
  MOV     PC, R14
It isn't a very good example, but you can imagine how it would be better to execute conditionally instead of branching. On the other hand, if you have large sections of code there or if your code uses the status flags, you can implement all sorts of branching using conditional execution: Thus the single simple branch instruction can replace all of those branch and jump instructions present in other processors. 
  OPT     1
  ADR     R1, #&70
  LDR     R0, [R1]
  CMP     R0, #0
  STRNE   R0, [R1, #2]
  MOV     PC, R14
  
  
[bookmark: bl]BL : Branch with Link
  BL<suffix>   <address>
BL is another branch instruction. This time, register 14 is loaded with the contents of R15 just before the branch. You can reload R14 into R15 to return to the instruction after the branch - a primitive but powerful implementation of a subroutine.
This effect is quite well shown up in the screen loader 2 (example 4)... 
    .load_new_format
      BL     switch_screen_mode
      BL     get_screen_info
      BL     load_palette

    .new_loop
      MOV    R1, R5
      BL     read_byte
      CMP    R0, #255
      BLEQ   read_loop
      STRB   R0, [R2, #1]!
...where we can see three subroutines are called before the loader loop. Then, the read_byte subroutine is called in the loop, once under conditional execution. 

Introduction
ARM processors after (and including) the ARM 3 offer various ID and internal configuration facilities by providing internally a co-processor 15 which you can read from and and write to. 
The setup is controlled by co-processor 15 registers, accessed with MRC and MCR in non-user mode. 
These registers are particular to the processor specified. 
  
  
ARM 3
· Register 0 - Processor identification (read only)
·   Bits  0 -  7  Revision of processor
·   Bits  8 - 15  Should be '3', identifying processor as an ARM3
·   Bits 16 - 23  Manufacturer code (&56 = VLSI Technology Inc.)
·   Bits 24 - 31  Designer code (&41 = ARM Ltd)

 
· Register 1 - Cache flush (write only)
Write-sensitive, writing anything to register 1 will cause the cache to be flushed. 
 
· Register 2 - Miscellaneous control
·   Bit 0 - Turns the cache on (1) or off (0)
·   Bit 1 - Determines if user mode and non-user mode use the same address
·           mapping. 1 if they do, or 0. Should be 1 for use with MEMC.
·   Bit 2 - 0 for normal operation, 1 for special monitor mode (processor
·           runs at memory speed and address/data always put on external
·           pins even if data fetched from cache - for logic analyser
·           to trace the program properly).
· 
·   Other bits reserved.

 
· Register 3 - Which areas are cachable
Controls which areas of memory are cachable, in 2Mb chunks. 
·   Bit 0  - 1 if virtual addresses &0000000-&01FFFFF are cachable, 0 if not
·   Bit 0  - 1 if virtual addresses &0200000-&03FFFFF are cachable, 0 if not
·   ...
·   Bit 31 - 1 if virtual addresses &3E00000-&3FFFFFF are cachable, 0 if not

 
· Register 4 - Which areas are updateable
Controls which areas of memory are updateable, in 2Mb chunks. Writes to non-updateable memory go to the real memory, not the cache. This is suitable for things like ROMs, since you don't want the cached data to be altered by attempted writes. 
·   Bit 0  - 1 if virtual addresses &0000000-&01FFFFF are updateable, 0 if not
·   Bit 0  - 1 if virtual addresses &0200000-&03FFFFF are updateable, 0 if not
·   ...
·   Bit 31 - 1 if virtual addresses &3E00000-&3FFFFFF are updateable, 0 if not

 
· Register 5 - Which areas are disruptive
Controls which areas of memory are disruptive, in 2Mb chunks. Writes to disruptive areas of memory cause the cache to be flushed. For example, writing to physical memory at &2000000-&2FFFFFF on an MEMC system will usually cache virtually addresses memory and if this location was cached, an attempt to read it would read back the old contents. 
·   Bit 0  - 1 if virtual addresses &0000000-&01FFFFF are disruptive, 0 if not
·   Bit 0  - 1 if virtual addresses &0200000-&03FFFFF are disruptive, 0 if not
·   ...
·   Bit 31 - 1 if virtual addresses &3E00000-&3FFFFFF are disruptive, 0 if not
Register 2 is set to zero after power-up, and registers 3-5 are undefined. The registers 3-5 should be set up correctly before the cache is switched on. You should always check the processoridentity before setting up the registers, unless you are completely certain your code will only ever be executed on an ARM3 processor. 
  
  
ARM 610
· Register 0 - Processor identification (read only)
The value returned for an ARM610 processor should be &4156061x.
·   Bits  0 -  7  Revision of processor (&1x)
·   Bits  8 - 15  Processor identity
·   Bits 16 - 23  Manufacturer code (&56 = VLSI Technology Inc.)
·   Bits 24 - 31  Designer code (&41 = ARM Ltd)

 
· Register 1 - Control (write only)
All values set to 0 at power-up.
·   Bit  0 - On-chip MMU turned off (0) or on (1)
·   Bit  1 - Address alignment fault disabled (0) or enabled (1)
·   Bit  2 - Instruction/data cache turned off (0) or on (1)
·   Bit  3 - Write buffer turned off (0) or on (1)
·   Bit  4 - 26 bit program space if 0, 32 bit program space if 1
·   Bit  5 - 26 bit data space if 0, 32 bit data space if 1
·   Bit  6 - Early abort mode if 0, late abort mode if 1
·   Bit  7 - Little-endian operation if 0, big-endian if 1
·   Bit  8 - System bit - controls the ARM610 permission system

 
· Register 2 - Translation Table Base (write only)
Bits 14-31 hold the base of the currently active Level One page table. 
 
· Register 3 - Domain Access Control (write only)
This register holds the current access control for domains 0 to 15. Each domain has two bits (domain 0 bits 0,1 ... domain 15 bits 30,31) which may be set as follows: 
·   00  No Access - Domain fault generated if tried to access
·   01  Client    - Accesses are checked against permission bits in
·                   section/page descriptor
·   10  Reserved  - Currently behaves like no access mode
·   11  Manager   - Accesses are NOT checked, permission faults cannot
·                   be generated

 
· Register 4 - Reserved - do not attempt to access 
 
· Register 5 - Page fault status / TLB flush
When reading, this holds the status of the last data fault (not updated for pre-fetch fault). Only the bottom byte is of significance. 
·   Bits  0 -  3  Status
·   Bits  4 -  7  Domain
·   Bits  8 - 11  Set to zero
·   Bits 12 - 31  Whatever was the last value on the internal data bus

 
When writing to this register, any value written will cause the Translation Look-aside Buffer to be flushed. 
 
· Register 6 - Data fault address / TLB purge
When reading this register, you can determine the virtual address of the last page fault. 
 
When writing this register, the value given (in bits 14-31) is treated as an address. The TLB will be searched for a corresponding address and if it is found, it is marked as invalid. This is to allow the page table in main memory to be updated and the now-invalid entries in the on-chip TLB to be purged without assuming the penalty of flushing the entire TLB. 
 
· Register 7 - IDC flush (write only)
Any data written to this location will cause the IDC (Instruction/Data cache) to be flushed. 
 
· Registers 8 to 15 - Reserved
Accessing these registers will cause the undefined instruction trap to be taken. 
  
  
ARM 710
This is similar to the ARM610. 
· Register 0 - Processor identification (read only)
The value returned for an ARM610 processor should be &4104710x.
·   Bits  0 -  3  Revision of processor?
·   Bits  3 - 15  Processor identity - &710
·   Bits 16 - 23  Manufacturer code
·   Bits 24 - 31  Designer code (&41 = ARM Ltd)

 
· Register 1 - Control (write only)
All values set to 0 at power-up.
·   Bit  0 - On-chip MMU turned off (0) or on (1)
·   Bit  1 - Address alignment fault disabled (0) or enabled (1)
·   Bit  2 - Instruction/data cache turned off (0) or on (1)
·   Bit  3 - Write buffer turned off (0) or on (1)
·   Bit  4 - 26 bit program space if 0, 32 bit program space if 1
·   Bit  5 - 26 bit data space if 0, 32 bit data space if 1
·   Bit  6 - Early abort mode if 0, late abort mode if 1
·   Bit  7 - Little-endian operation if 0, big-endian if 1
·   Bit  8 - System bit - controls the ARM710 permission system
·   Bit  9 - ROM bit - controls the ARM710 permission system

 
· Register 2 - Translation Table Base (write only)
Bits 14-31 hold the base of the currently active Level One page table. 
 
· Register 3 - Domain Access Control (write only)
This register holds the current access control for domains 0 to 15. Each domain has two bits (domain 0 bits 0,1 ... domain 15 bits 30,31) which may be set as follows: 
·   00  No Access - Domain fault generated if tried to access
·   01  Client    - Accesses are checked against permission bits in
·                   section/page descriptor
·   10  Reserved  - Currently behaves like no access mode
·   11  Manager   - Accesses are NOT checked, permission faults cannot
·                   be generated

 
· Register 4 - Reserved - do not attempt to access 
 
· Register 5 - Page fault status / TLB flush
When reading, this holds the status of the last data fault (not updated for pre-fetch fault). Only the bottom byte is of significance. 
·   Bits  0 -  3  Status
·   Bits  4 -  7  Domain
·   Bits  8 - 11  Set to zero
·   Bits 12 - 31  Whatever was the last value on the internal data bus

 
When writing to this register, any value written will cause the Translation Look-aside Buffer to be flushed. 
 
· Register 6 - Data fault address / TLB purge
When reading this register, you can determine the virtual address of the last page fault. 
 
When writing this register, the value given (in bits 14-31) is treated as an address. The TLB will be searched for a corresponding address and if it is found, it is marked as invalid. This is to allow the page table in main memory to be updated and the now-invalid entries in the on-chip TLB to be purged without assuming the penalty of flushing the entire TLB. 
 
· Register 7 - IDC flush (write only)
Any data written to this location will cause the IDC (Instruction/Data cache) to be flushed. 
 
· Registers 8 to 15 - Reserved
Accessing these registers will cause the undefined instruction trap to be taken. 
  
  
ARM 7500
The registers are exactly the same as the ARM710, except the processor ID (register 0) will be different. The datasheet did not specify what should be expected. 
  
  
ARM 7500FE
The registers are exactly the same as the ARM710, except the processor ID (register 0) will be different. The datasheet did not specify what should be expected, however interrogation of the Bush set-top box reveals &41077100. 
  
  
StrongARM SA110
· Register 0 - Processor identification (read only)
The value returned for an SA110 processor should be &4401A10x.
  Bits  0 -  3  Processor revision number

 
· Register 1 - Control (read/write)
All values set to 0 at power-up.
·   Bit  0 - On-chip MMU turned off (0) or on (1)
·   Bit  1 - Address alignment fault disabled (0) or enabled (1)
·   Bit  2 - Data cache turned off (0) or on (1)
·   Bit  3 - Write buffer turned off (0) or on (1)
·   Bit  7 - Little-endian operation if 0, big-endian if 1
·   Bit  8 - System bit - controls the MMU permission system
·   Bit  9 - ROM bit - controls the MMU permission system
·   Bit 12 - Instruction cache turned off (0) or on (1)

 
· Register 2 - Translation Table Base (read/write)
Bits 14-31 hold the base of the currently active Level One page table. 
 
· Register 3 - Domain Access Control (read/write)
This register holds the current access control for domains 0 to 15.
The document I have contains no further details, though I would assume it would be similar to the ARM610/710/etc usage. 
 
· Register 4 - Reserved - do not attempt to access 
 
· Register 5 - Fault status (read/write)
When reading, this holds the status of the last data fault (not updated for pre-fetch fault). Only the bottom byte is of significance. 
·   Bits  0 -  3  Status
·   Bits  4 -  7  Domain
·   Bit        8  Zero
·   Bits  9 - 31  Undefined on read, ignored on write

 
· Register 6 - Fault address (read/write)
When reading this register, you can determine the virtual address of the last page fault. 
 
· Register 7 - Cache control (write only)
Any data written to this location will cause the selected cache to be flushed. 
·   The OPC_2 and CRm co-processor fields select which cache
·   operation should occur:
· 
·     Function         OPC_2    CRm    Data
· 
·     Flush I + D      %0000    %0111  -
·     Flush I          %0000    %0101  -
·     Flush D          %0000    %0110  -
·     Flush D single   %0001    %0110  Virtual address
·     Clean D entry    %0001    %1010  Virtual address
·     Drain write buf. %0100    %1010  -

 
· Register 8 - TLB operations (write only)
Any data written to this location will cause the selected TLB flush operation. 
·   The OPC_2 and CRm co-processor fields select which cache
·   operation should occur:
· 
·     Function         OPC_2    CRm    Data
· 
·     Flush I + D      %0000    %0111  -
·     Flush I          %0000    %0101  -
·     Flush D          %0000    %0110  -
·     Flush D single   %0001    %0110  Virtual address

 
· Registers 9 to 14 - Reserved
Accessing these registers will cause the undefined instruction trap to be taken. 
 
· Register 15 - Test, Clock, and Idle (write only)
·   The OPC_2 and CRm co-processor fields select the following...
· 
·     Function         OPC_2    CRm
· 
·     Enable odd word  %0001    %0001
·     loading of
·     Icache LFSR
· 
·     Enable even word %0001    %0010
·     loading of
·     Icache LFSR
· 
·     Clear Icache     %0001    %0100
·     LFSR
· 
·     Move LFSR to     %0001    %1000
·     R14,Abort
· 
·     Enable clock     %0010    %0001
·     switching
· 
·     Disable clock    %0010    %0010
·     switching
· 
·     Disable nMCLK    %0010    %0100
·     output
· 
·     Wait for         %0010    %1000
·     interrupt
  
  
ARM9...XScale
Unfortunately I do not have details of these registers.
Try http://www.arm.com/. 
  
  
How to read these registers
The code I knocked up for the Bush box processor ID was: 
    10 DIM code% 32
    20 P% = code%
    30 [ OPT     3
    40   SWI     "OS_EnterOS"
    50   MRC     CP15, 0, R0, C0, C0
    60   TSTP    PC, #&F0000000
    70   MOV     R0, R0
    80   MOV     PC, R14
    90 ]
   100 PRINT ~USR(code%)
When run, this would print: 
   >RUN
   00008FAC                    OPT     3
   00008FAC EF000016           SWI     "OS_EnterOS"
   00008FB0 EE100F10           MRC     CP15, 0, R0, C0, C0
   00008FB4 E31FF20F           TSTP    PC, #&F0000000
   00008FB8 E1A00000           MOV     R0, R0
   00008FBC E1A0F00E           MOV     PC, R14
     41077100
   >
Note that this code must run in a privileged mode. 
  
  
[bookmark: copro]Co-processors
There are between zero and three possible co-processors. Most desktop ARM systems do not have logic for external co-processors, so we may either use that which is built into the ARM itself, or an emulated co-processor.
CP15 is reserved on the ARM 3 and later processors for internal configuration, as described in this document.
CP0 and CP1 is used by the floating point system. It may either be an external floating point chip (as used with the ARM 3), hardware built into the processor (as in the ARM 7500FE), or a totally software-based emulation (as with the FPEmulator that we all know). 
Here is a short exercise for you: 
    10 DIM code% 16
    20 P% = code%
    30 [ OPT     3
    40   CDP     CP1, 0, C0, C1, C2, 0
    50   ADFS    F0, F1, F3
    60   MOV     PC, R14
    70 ]
   >RUN
   00008F78                    OPT     3
   00008F78 EE010102           CDP     CP1, 0, C0, C1, C2
   00008F7C EE010102           ADFS    F0, F1, F2
   00008F80 E1A0F00E           MOV     PC, R14
   >
What do you notice? :-) 
  
When the ARM executes a co-processor instruction, or an undefined instruction, it will offer it to any co-processors which may be presently attached. If hardware is available to process the given instruction, then it is expected to do so. If it is busy at the time the instruction is offered, the ARM will wait for it.
If there is no co-processor capable of executing the instruction, the ARM will take its undefined instruction trap, in which case the following will happen: 
· The PSR and PC are both saved (the method differs for 26 bit and 32 bit ARMs) 
· SVC mode (26 bit) / UND mode (32 bit) is entered, and the I bit of the PSR is set 
· The instruction at address &00000004 is executed 
This trap may be used to add instructions to the instruction set by emulation, or to implement a software emulation of hardware that isn't fitted. The Floating Point Emulator works by doing this. 
To return, simply pull the saved PC and PSR (depends on 26/32 bit) and push them to the current PC and PSR, like MOVS PC, R14 in 26 bit systems. This will pick up with the instruction following the one which caused the trap. 
All of the co-processor instructions can be executed conditionally. Please note that the conditionals relate to the status of the ARM processor, and not the status of any of the co-processors. This is because the ARM always tries the instruction first, and offers it around and maybe takes the undefined application trap, so the conditions are ARM related.
To make this clearer: 
    10 DIM code% 32
    20 P% = code%
    30 [ OPT     3
    40   FLTS    F0, R0
    50   FLTS    F1, R1
    60   FMLS    F2, F0, F1
    70   FIX     R0, F2
    80   MOVS    PC, R14
    90 ]
   100 INPUT "First number : "A%
   110 INPUT "Second number: "B%
   120 PRINT USR(code%)
This probably won't assemble without an enhanced BASIC assembler. 
Anyway, you might think the ARM will hand over to the floating point co-processor to do the four FP instructions, then hand back afterwards.
If you did, you would be incorrect! 
What actually is executed is: 
   MCR     CP1, 0, R0, C0, C0
   MCR     CP1, 0, R1, C1, C0
   CDP     CP1, 9, C2, C0, C1
   MRC     CP1, 0, R0, C0, C2
It is worth pointing out that objasm specifies co-processor registers using the CR notation (ie, CR0 - CR15), which is first defined with the CN directive. It does not appear as if default co-processor instructions are defined in Nick Roberts' ASM, though I've only looked in the instructions at the "defined symbols" section...
Darren Salt's ExtBASICasm provides the register names C0 - C15 to refer to the co-processors. So if any of these examples fail when you try to assemble them, please check what format your assembler provides these instructions. 
  
  
[bookmark: mrc]MRC
The instruction MRC transfers a co-processor register to an ARM register. It takes the form: 
   MRC    <co-pro>, <op>, <ARM reg>, <co-pro reg>, <co-pro reg2>, <op2>
The co-processor is denoted in most assemblers by CPx.
The register <co-pro reg> is written to <ARM reg>, using operation <op>. This may, possibly, be further modified by <co-pro reg2> and <op2>. For an idea of the sorts of times when this might be necessary, consider instructions of the form LDR Ra, [Rb], #x. 
The final <op2> may be omitted, as it is in the example, but the other parts of the MRC instruction must be supplied. 
  
[bookmark: mcr]MCR
The instruction MCR transfers an ARM register to a co-processor register. It takes the form: 
   MCR    <co-pro>, <op>, <ARM reg>, <co-pro reg>, <co-pro reg2>, <op2>
The co-processor is free to interpret the fields as it desires, but the standard interpretation is that the contents of the ARM register are written to the co-processor register using the operation code given, which may be further modified by the second co-processor register and/or the second operation code. 
  
[bookmark: ldc][bookmark: stc]LDC and STC
The instruction LDC loads data from memory into the co-processor register, while STC saves data from a co-processor register to memory.
The ARM should supply the address, the co-processor accepts the data and controls how much is transferred. 
   LDC    <co-pro>, <co-pro reg>, <address>
   LDCL   <co-pro>, <co-pro reg>, <address>
   STC    <co-pro>, <co-pro reg>, <address>
   STCL   <co-pro>, <co-pro reg>, <address>
If the 'L' flag is specified, a long transfer is performed. Otherwise a short transfer is performed. The 'L' flag follows the extension, like LDCEQL.
The address is an expression which results in an address being generated, so examples of which are: 
   [Rx]
   [Rx, #x] !
   [Rx], #x
These are like those used for the LDR instruction. However they are only eight bits wide and specify word offsets (the ARM types are 12 bit and byte offset).
What happens is the 8 bit unsigned offset is shifted left two bits and added or subtracted from the base register, this may be done before or after the base is used as the transfer address. The new base value can be written back, or left unmodified.
The next difference is that post-indexed addressing requires explicit setting of the W bit of the instruction (unlike LDR/STR which always does it when post-indexed). You set the 'W' bit with the '!' flag, like STR CP0, CR1, [R2, #16]!.
The base register is used for the first transfer. If there are any further transfers, the base will be incremented by one word for each of those additional transfers. 
  
[bookmark: cdp]CDP
The instruction CDP instructs the co-processor to do some processing. It takes the form: 
   CDP    <co-pro>, <co-pro reg1>, <co-pro reg2>, <co-pro reg3>, <op>
This tells the co-processor to do something. The ARM will not wait for it to finish, nor is any sort of status sent back to the ARM. It is possible for a co-processor to maintain a queue of instructions, allowing it and the ARM to process in parallel.
A variant of this may be obtained with the floating point hardware; while it does not (to my knowledge) support a queue of instructions, it is true that the ARM will await the FPU to finish an operation before providing the next. With careful coding, it would therefore be possible to get the ARM to do some sort of processing (a few instructions) in between sending an instruction to the FPU and reading it's result back.
So instead of: 
   FLTE    F0, R0
   FLTE    F1, R1
   MUFE    F2, F0, F1
   FIX     R0, F2
   MOV     R1, #0
you could save a small amount of time with: 
   FLTE    F0, R0
   FLTE    F1, R1
   MUFE    F2, F0, F1
   MOV     R1, #0
   FIX     R0, F2
as the FPU could be finishing the MUF while you MOV. The hardware FPU (as in the 7500FE) runs asynchronous - you can switch to synchronous by setting a bit in the FPSR. The software emulation always runs synchronously, and as it uses the ARM in order to emulate the FP instructions, there is no possible advantage to be gained.
Obviously the above example is somewhat contrived. However it is only an example. Real life code, such an an MP3 decoder, could well benefit from careful arrangement of code. 
There are no rules for the register types and/or the operation codes. These depend upon the co-processor. 

These are, arguably, the most useful instructions available. It is all very well being able to do stuff with the registers, but if you cannot load and store them to the main memory, then... <grin> 
  
[bookmark: str][bookmark: ldr]Single Data Transfer
The single data transfer instructions (STR and LDR) are used to load and store single bytes or words of data from/to main memory. The addressing is very flexible. 
First, we'll look at the instruction: 
  LDR    R0, address
  STR    R0, address
  LDRB   R0, address
  STRB   R0, address
These instructions load and store the value of R0 to the specified address. If 'B' is also specified, as in the latter two instructions, then only a single byte is loaded or saved. The three unused bytes in the word are zeroed upon loading. 
The address can be a simple value, or an offset, or a shifted offset. Write-back may be performed (to remove the need for adding/subtracting). 
   STR    R0, [Rbase]          Store R0 at Rbase.

   STR    R0, [Rbase, Rindex]  Store R0 at Rbase + Rindex.

   STR    R0, [Rbase, #index]  Store R0 at Rbase + index.
                               Index is an immediate value.
                               STR R0, [R1, #16] would load R0
                               from R1+16.

   STR    R0, [Rbase, Rindex]! Store R0 at Rbase + Rindex, &
                               write back new address to Rbase.

   STR    R0, [Rbase, #index]! Store R0 at Rbase + index, &
                               write back new address to Rbase.

   STR    R0, [Rbase], Rindex  Store R0 at Rbase, & write back
                               Rbase + Rindex to Rbase.

   STR    R0, [Rbase, Rindex, LSL #2] will store R0 at the address
                               Rbase + (Rindex * 4)

   STR    R0, place            Will generate a PC-relative offset
                               to 'place', and store R0 there.
You can, of course, use conditional execution on any of these instructions. Note, however, that the conditional flag comes before the byte flag, so if you wish to load a byte when the result is equal, the instruction would be LDREQB Rx, address (not LDRBEQ...). 
If you specify pre-indexed addressing (where the base and index are both within square brackets), the write-back is controlled by the presence or absence of the '!'. The fourth and fifth examples above reflect this. Using this, you can automatically move forward or backward in memory. A string print routine could then become: 
  .loop
    LDRB   R0, [R1, #1]!
    SWI    "OS_WriteC"
    CMP    R0, #0
    BNE    loop
instead of: 
  .loop
    LDRB   R0, [R1]
    SWI    "OS_WriteC"
    ADD    R1, R1, #1
    CMP    R0, #0
    BNE    loop
The use of '!' is invalid for post-indexed addressing (where the index is outside of the square brackets, as in example six above) as write-back is implied. 
As you can see, the offset may be shifted. Additionally, the index offset may be subtracted from the base. In this case, you might use code such as: 
  LDRB   R0, [R1, #-1]
You cannot modify the PSR with a load or store instruction, though you can store or load the PC. In order to load a stored 'state' and correctly restore it, use: 
  LDR    R0, [Rbase]
  MOVS   R15, R0
The MOVS will cause the PSR bits to be updated, provided that you are privileged.
Using MOVS with PC is not 32-bit compliant. 
According to the ARM assembler manual:
A byte load (LDRB) expects the data on bits 0 to 7 if the supplied address is on a word boundary, on bits 8 to 15 if it is a word address plus one byte, and so on. The selected byte is placed in the bottom 8 bits of the destination register, and the remaining bits of the register are filled with zeroes.
A byte store (STRB) repeats the bottom 8 bits of the source register four times across the data bus. The external memory system should activate the appropriate byte subsystem to store the data.
A word load (LDR) or word store (STR) should generate a word aligned address. Using a non-word-aligned addresses has non-obvious and unspecified results. 
The only thing of real note here is that you cannot use LDR to load a word from a non-aligned address. 
  
[bookmark: stm]Multiple Data Transfer
The multiple data transfer instructions (LDM and STM) are used to load and store multiple words of data from/to main memory. 
The main use of LDM/STM is to dump registers that need to be preserved onto the stack. We've all seen STMFD R13!, {R0-R12, R14}. 
The instruction is: 
  xxM type cond   base write-back, {register list}
'xx' is LD to load, or ST to store. 
'type' is: 
  Stack     Other
  LDMED     LDMIB     Pre-incremental load
  LDMFD     LDMIA     Post-incremental load
  LDMEA     LDMDB     Pre-decremental load
  LDMFA     LDMDA     Post-decremental load 

  STMFA     STMIB     Pre-incremental store
  STMEA     STMIA     Post-incremental store
  STMFD     STMDB     Pre-decremental store
  STMED     STMDA     Post-decremental store
The assembler takes care of how to map the mnemonics. Note that ED is not IB; it is only the same for a pre-decremental load. When storing, ED is post-decrement. 
FD, ED, FA, and EA refer to a Full or Empty stack which is either Ascending or Descending.
A full stack is where the stack pointer points to the last data item written, and empty stack is where the stack pointer points to the first free slot.
A descending stack grows downwards in memory (ie, from the end of application space down) and an ascending stack is one which grows upwards in memory. 
The other forms simply describe the behaviour of the instruction, and mean Increment After, Increment Before, Decrement After, Decrement Before. 
RISC OS, by tradition, uses a Fully Descending stack. When writing in APCS assembler, it is common to set your stack pointer to the end of application space and then use a Full Descending stack. If you are working with a high level language (either BASIC or C), then you don't get a choice. The stack pointer (traditionally R13) points to the end of a fully descending stack. You must continue this format, or create and manage your own stack (if you're the sort of die-hard person that would do something like this!). 
'base' is the register containing the address to begin with. Traditionally under RISC OS, the stack pointer is R13, though you can use any available register except R15. 
If you would like the stack pointer to be updated with the new register contents, simply set the write-back bit by following the stack pointer register with an '!'. 
The register list is given in {curly brackets}. It doesn't matter what order you specify the registers in, they are stored from lowest to highest. As a single bit determines whether or not a register is saved, there is no point to trying to specify it twice.
A side effect of this is that code such as: 
  STMFD  R13!, {R0, R1}
  LDMFD  R13!, {R1, R0}
will not swap the contents of two registers.
A useful shorthand has been provided. To encompass a range of registers, simply say the first and the last, and put a dash between them. For example R0-R3 is identical to R0, R1, R2, R3, only tidier and saner... 
When R15 is stored to memory, the PSR bits are also saved. When R15 is reloaded, the PSR bits are NOT restored unless you request it. The method of requesting is to follow the register list with a '^'. 
  STMFD  R13!, {R0-R12, R14}
  ...
  LDMFD  R13!, {R0-R12, PC}
This saves all registers, does some stuff, then reloads all registers. PC is loaded from R14 which was probably set by a BL instruction or some-such. The PSR flags are untouched. 
  STMFD  R13!, {R0-R12, R14}
  ...
  LDMFD  R13!, {R0-R12, PC}^
This saves all registers, does some stuff, then reloads all registers. PC is loaded from R14 which was probably set by a BL instruction. The PSR flags are updated.
Warning: This code is not 32 bit compliant. You need to use MRS and MSR to handle the PSR. You cannot use the '^' suffix. 
Note that in both examples, R14 is loaded directly into PC. This saves the need to MOV(S) R14 into R15.
Warning: Using MOVS PC, ... is not 32 bit compliant. You need to use MRS and MSR to handle the PSR.

These two instructions are different from the normal arithmetical instructions in that there are restrictions on the operands, namely: 
1. All operands, and the destination, must be given as simple registers. 
2. You cannot use immediate values or shifted registers for operand two. 
3. The destination and operand one must be different registers. 
4. Lastly, you cannot specify R15 as the destination. 
[bookmark: mla]MLA : Multiplication with Accumulate
  MLA<suffix>  <dest>, <op 1>, <op 2>, <op 3>

                dest = (op_1 * op_2) + op_3
MLA behaved that same as MUL, except that the value of operand three is added to the result. This is useful for running totals. 
  
  
[bookmark: mul]MUL : Multiplication
  MUL<suffix>  <dest>, <op 1>, <op 2>

                dest = op_1 * op_2
MUL provides 32 bit integer multiplication. If the operands are signed, it can be assumed that the result is also signed.
Here is an example of multiplication: 
REM Multiplication example
REM
REM by Richard Murray 26th April 1999
REM
REM Downloaded from: http://www.heyrick.co.uk/assembler/

DIM code% 12

P%=code%
[  OPT 2
   MUL     R2, R0, R1
   MOV     R0, R2
   MOV     PC, R14
]

REPEAT
  INPUT "Number 1 : "A%
  INPUT "Number 2 : "B%
  PRINT "Result   : "+STR$(USR(code%))'
UNTIL A% = 0

END

Go here for details of R15/PSR in 32-bit mode. 
Register 15 (26-bit mode):
R15 is built up as follows: 
  Bit  31  30  29  28  27  26  25------------2  1  0

       N   Z   C   V   I   F   Program Counter  S1 S0
The flags mean: 
  N  Negative        Set if result is negative
  Z  Zero            Set if result is zero
  C  Carry           Set if carry occurs
  O  Overflow        Set if overflow occurs
  I  IRQ             Interrupt disable
  F  FIQ             Fast Interrupt disable
  
  S1 and S0 are processor mode flags:
  
           S1   S0   Mode
           0    0    USR - User mode
           0    1    FIQ - Fast Interrupt mode
           1    0    IRQ - Interrupt mode
           1    1    SVC - Supervisor mode
When R15 is used as the first operand in an instruction, only the Program Counter part of it is available. Thus, the following instruction will copy PC out to a register and add 256 to it:
  ADD    R0, R15, #256
(R15 and PC mean the same thing to the BASIC assembler) 
When R15 is used as the second operand, all 32 bits are accessible: the Program Counter, the flags, and the status. The following code segment will identify the current processor mode: 
   MOV     R0, #3          ; Load a bit mask (%11) into R0
   AND     R0, R0, PC      ; AND R15 into R0, to get the mode status
   CMP     R0, #3          ; Compare mode with '3' (SVC)
   BEQ     svc             ; If SVC mode, branch to 'svc'
   CMP     R0, #2          ; Compare mode with '2' (IRQ)
   BEQ     irq             ; If IRQ mode, branch to 'irq'
   CMP     R0, #1          ; Compare mode with '1' (FIQ)
   BEQ     fiq             ; If FIQ mode, branch to 'fiq'
   CMP     R0, #0          ; Compare mode with '0' (USR)
   BEQ     usr             ; If USR mode, branch to 'usr'
This example is not 32-bit compliant. Refer to the section below for how to read the current mode in the 32-bit environment.
Download example: currmode.basic
Changing processor status:
In order to change the processor mode, or indeed any of the flags, we need to EOR the desired flag with the status flags,
  new_state = old_state EOR (1 << 28) could be pseudocode for changing the state of the oVerflow flag. But we cannot do a simple EORS operation as the pipeline would cause the following two instructions to be skipped.
But don't worry. The instruction TEQ does a pretend EOR (the results are not stored anywhere). Combine this with the P suffix, which writes bits 0, 1, and 26 to 31 of the result directly to bits 0, 1, and 26 to 31 of R15 giving you an easy way to change the flags:   TEQP   R15, bit_mask 
You can only change a flag if you are in a processor mode which allows you to set that flag.
This example is not 32-bit compliant. Refer to the section below for how to change modes in the 32-bit environment. 
This can be expanded to change processor mode. For example, to enter SVC mode you would: 
   MOV     R6, PC          ; Store original state of PC in R6
   ORR     R7, R6, #3      ; Set SVC mode
   TEQP    R7, #0          ; Write mode flags (in R7) to PC
And to return to the original mode: 
   TEQP    R6, #0          ; Write previous mode flags (in R6) to PC
Download example: setmode.basic
After changing the mode, you should perform a null operation to allow the registers to settle. Something like MOV R0, R0 should be okay. The use of NV suffixed instructions has been deprecated. 
  
  
Conditional execution:
A very special feature of the ARM processor is its conditional execution. We are not talking your basic Branch if Carry Set, the ARM takes this a logical stage further to mean XXX if carry set - where XXX is just about anything. 
By way of example, here is a list of branch instructions understood by the Intel 8086 processor: 
  JA    Jump if Above
  JAE   Jump if Above or Equal
  JB    Jump if Below
  JBE   Jump if Below or Equal
  JC    Jump if Carry
  JCXZ  Jump if CX Zero (CX is a register that can be used for loop counts)
  JE    Jump if Equal
  JG    Jump if Greater than
  JGE   Jump if Greater than or Equal
  JL    Jump if Less than
  JLE   Jump if Less Than or Equal
  JMP   JuMP (unconditional)
  JNA   Jump if Not Above
  JNAE  Jump if Not Above or Equal
  JNB   Jump if Not Below
  JNBE  Jump if Not Below or Equal
  JNC   Jump if No Carry
  JNE   Jump if Not Equal
  JNG   Jump if Not Greater than
  JNGE  Jump if Not Greater than or Equal
  JNL   Jump if Not Less than
  JNLE  Jump if Not Less than or Equal
  JNO   Jump if Not Overflow
  JNP   Jump if Not Parity
  JNS   Jump if Not Sign
  JNZ   Jump if Not Zero
  JO    Jump if Overflow
  JP    Jump if Parity
  JPE   Jump if Parity Even
  JPO   Jump if Parity Odd
  JS    Jump if Sign
  JZ    Jump if Zero

And the 80386 added:
  JCXZ  Jump if CX Zero
  JECXZ Jump if ECX Zero

Not forgetting the closely related:
  LOOP          Decrement CX and loop if not zero
  LOOPE/LOOPZ   Decrement (E)CX and loop if not zero and ZF = 1
  LOOPNE/LOOPNZ Decrement (E)CX and loop if not zero and ZF = 0


Added caveat:

  JZ    far_address

is illegal, you must use:

  JNZ   skip_jump
  JMP   far_address
  skip_jump:

which messes up the pipeline as only JMP can jump to a 'far' address.
And by contrast, the ARM processor offers a whopping...uh... 
  B     Branch
  BL    Branch with Link
  
But the ARM is not limited by this seemingly inflexible approach due to conditional execution which offers you: 
  BEQ   Branch if EQual
  BNE   Branch if Not Equal
  BVS   Branch if oVerflow Set
  BVC   Branch if oVerflow Clear
  BHI   Branch if HIgher
  BLS   Branch if Lower or the Same
  BPL   Branch if PLus
  BMI   Branch if MInus
  BCS   Branch if Carry Set
  BCC   Branch if Carry Clear
  BGE   Branch if Greater than or Equal
  BGT   Branch if Greater Than
  BLE   Branch if Less than or Equal
  BLT   Branch if Less Than

  BLEQ  Branch with Link if EQual
  ....
  BLLT  Branch with Link if Less Than
There are two more codes, 
· AL - ALways, the default condition so doesn't need to be specified 
· NV - NeVer, so very useful. You should not use this code anyway... 
The crunch comes, however, when you realise that all of the Bxx instructions are actually the same instruction. Then you will think, if you can do all that to a branch instruction, can it be done to, say, a register load instruction?
The answer is yes. 
  
  
Here follows a list of available conditional codes: 
EQ : Equal 
If the Z flag is set after a comparison. 
NE : Not Equal 
If the Z flag is clear after a comparison. 
VS : Overflow Set 
If the V flag is set after an arithmetical operation, the result of which will not fit into a 32bit destination register. 
VC : Overflow Clear 
If the V flag is clear, the reverse of VS. 
HI : Higher Than (unsigned) 
If after a comparison the C flag is set AND the Z flag is clear. 
LS : Lower Than or Same (unsigned) 
If after a comparison the C flag is clear OR the Z flag is set. 
PL : Plus 
If the N flag is clear after an arithmetical operation. For the purposes of defining 'plus', zero is positive because it isn't negative... 
MI : Minus 
If the N flag is set after an arithmetical operation. 
CS : Carry Set 
Set if the C flag is set after an arithmetical operation OR a shift operation, the result of which cannot be represented in 32bits. You can think of the C flag as the 33rd bit of the result. 
CC : Carry Clear 
The reverse of CS. 
GE : Greater Than or Equal (signed) 
If after a comparison...
the N flag is set AND the V flag is set
or...
the N flag is clear AND the V flag is clear. 
GT : Greater Than (signed) 
If after a comparison...
the N flag is set AND the V flag is set
or...
the N flag is clear AND the V flag is clear
and...
the Z flag is clear. 
LE : Less Than or Equal To (signed) 
If after a comparison...
the N flag is set AND the V flag is clear
or...
the N flag is clear AND the V flag is set
and...
the Z flag is set. 
LT : Less Than (signed) 
If after a comparison...
the N flag is set AND the V flag is clear
or...
the N flag is clear AND the V flag is set. 
AL : Always 
The default condition, so does not need to be explicity stated. 
NV : Never 
Not particularly useful, it states that the instruction should never be executed. A kind of Poor Mans' NOP.
NV was included for completeness (as the reverse of AL), but you should not use it in your own code. 
There is a final conditional code which works in the reverse way. S, when applied to an instruction, causes the status flags to be updated. This does not happen automatically - except for those instructions whose purpose is to set the status. For example: 
  ADD     R0, R0, R1

  ADDS    R0, R0, R1

  ADDEQS  R0, R0, R1
The first example shows us a basic addition (adding the value of R1 to R0) which does not affect the status registers. 
The second example shows us the same addition, only this time it will cause the status registers to be updated. 
The last example shows us the addition again, updating the status registers. The difference here is that it is a conditional instruction. It will only be executed if the result of a previous operation was EQual (if the Z flag is set). 
  
  
Here is an example of conditional execution at work. You want to compare register zero with the contents of something stored in register ten. If not equal to R10, then call a software interrupt, increment and branch back to do it again. Otherwise clear R10 and return to a calling piece of code (whose address is stored in R14). 

  \ An example of conditional execution

  .loop                           ; Mark the loop start position
  CMP     R0, R10                 ; Compare R0 with R10
  SWINE   &40017                  ; Not equal: Call SWI &40017
  ADDNE   R0, R0, #1              ;            Add 1 to R0
  BNE     loop                    ;            Branch to 'loop'
  MOV     R10, #0                 ; Equal    : Set R10 to zero
  LDMFD   R13!, {R0-R12,PC}       ;            Return to caller
Notes: 
· The SWI number was made up as I was writing this. Under RISC OS, it is the number for Econet_DoImmediate. Don't take it literally, its only an example! 
· If you've not see LDMFD before, it loads multiple registers from the stack. In this example, we are loading R0 to R12 and R14 from a fully descending stack. Refer to str.html for more on register loading and storing. 
· Yes, I did just say we load R14. So why put it in the PC? Because the value of R14 when stored contains the return address. We can give happily:
LDMFD   R13!, {R0-R12,R14}
MOV     PC, R14
However by reloading straight into PC, we can omit the MOV statement. 
· Finally, it is very likely that registers will be corrupted by a SWI call (dependant upon the code executed during the call), so you might like to push the important registers on the stack, to restore them later. 
  
  
[bookmark: 32bit]The 32 bit PSR
As described in this document, processors after the ARM 3 provide a 32 bit addressing space by moving the PSR out of R15 and giving R15 a full 32 bits in which to store the address of the current position.
Currently, RISC OS works in 26 bit mode, except for a few special cases which is unlikely to be encountered.
The 32 bit mode is important because 26 bits (as in the old PSR) restricts the maximum amount of addressable memory per-application to 28Mb. That is why you can't drag the Next slot beyond 28Mb irrespective of how much memory you have installed. 
The allocation of the bits within the CPSR (and the SPSR registers to which it is saved) is: 
  31 30 29 28  ---   7   6   -   4   3   2   1   0
  N  Z  C  V         I   F       M4  M3  M2  M1  M0

                                 0   0   0   0   0     User26 mode
                                 0   0   0   0   1     FIQ26 mode
                                 0   0   0   1   0     IRQ26 mode
                                 0   0   0   1   1     SVC26 mode
                                 1   0   0   0   0     User mode
                                 1   0   0   0   1     FIQ mode
                                 1   0   0   1   0     IRQ mode
                                 1   0   0   1   1     SVC mode
                                 1   0   1   1   1     ABT mode
                                 1   1   0   1   1     UND mode
  
Typically, the processor will be operating in User26, FIQ26, IRQ26 or SVC26 mode. It is possible to enter a 32 bit mode, but extreme care must be taken. RISC OS won't expect it, and will sulk greatly if it finds itself in it!
(except RISC OS 5 which works totally in 32bit mode - and you cannot enter 26bit as the processor doesn't have that anymore...) 
You cannot use MOVS PC, R14 or LDMFD R13!, {registers, PC}^ in 32 bit code. Neither can you use ORRS PC, R14, #1<<28 to set the V flag.
All of this is now possible using MRS and MSR. 
Copy a register into the PSR
  MSR     CPSR, R0                ; Copy R0 into CPSR
  MSR     SPSR, R0                ; Copy R0 into SPSR
  MSR     CPSR_flg, R0            ; Copy flag bits of R0 into CPSR
  MSR     CPSR_flg, #1<<28        ; Copy flag bits (immediate) into CPSR
Copy the PSR into a register
  MRS     R0, CPSR                ; Copy CPSR into R0
  MRS     R0, SPSR                ; Copy SPSR into R0
You have two PSRs - CPSR which is the Current Program Status Register and SPSR which is the Saved Program Status Register (the previous processor mode's PSR). 
Each privileged mode has its own PSR, so the total available selection of PSR is: 
· CPSR_all - current 
· SPSR_svc - saved, SVC(32) mode 
· SPSR_irq - saved, IRQ(32) mode 
· SPSR_abt - saved, ABT(32) mode 
· SPSR_und - saved, UND(32) mode 
· SPSR_fiq - saved, FIQ(32) mode 
It appears as if you cannot explicitly specify to save the current PSR in, say, SPSR_fiq. Instead, you should change to FIQ mode and then save to SPSR. In other words, you can only alter the SPSR of the mode you are in.
Using the _flg suffix allows you to alter the flag bits without affecting the control bits. 
In user(32) mode, the control bits of CPSR are protected, you can only alter the condition flags. In other modes, the entire CPSR is available. You should not specify R15 as a source or destination register. And finally, you must not attempt to access the SPSR in user(32) mode as it doesn't exist! 
To set the V flag: 
  MSR     CPSR_flg, #&10000000
This sets the V flag and doesn't affect the control bits. 
Here, for your delectation, is a way to set the V flag on any ARM processor: 
  CMP     R0, #1<<31
  CMNVC   R0, #1<<31
Clever, huh? 
To change mode: 
  MRS     R0, CPSR_all            ; Copy the PSR
  BIC     R0, R0, #&1F            ; Clear the mode bits
  ORR     R0, R0, #new_mode       ; Set bits for new mode
  MSR     CPSR_all, R0            ; write PSR back, changing mode
  
IMPORTANT: Your assembler may complain that 'CPSR_all' is not recognised, or that you need some weird syntax such as 'CPSR_nvf'; refer to the documentation supplied with your assembler to figure out exactly what it's MSR/MRS syntax is... 
  
What we shall do now is to enter the SVC32 mode and set the Z flag. Then we'll return to SVC26 mode and 'test' if the Z flag is set.
RISC OS will not be expecting to find itself in 32 bit mode, so we'll disable all interrupts and keep them that way. Though this code should execute fast enough, and we don't want to take any chances... 
Before we begin, I just want to say that you will need Darren Salt's ExtBASICasm module to allow BASIC to assemble the MSR/MRS instructions. It is entirely possible to insert these instructions using EQUD, but it is messy and hard to follow. 
If you use !Zap, you'll find it as !Zap.Code.Extensions.ExtAsm, in the v1.40 distribution. 

REM >32bittest
REM
REM Short example to go into 32 bit mode, set the
REM Z flag, and return to a conditional based upon
REM this.
REM
REM Downloaded from http://www.heyrick.co.uk/assembler/
REM

REM We need to ensure the required hardware is present...
emsg$ = "Sorry, you'll need a RiscPC or later, with a 32 bit capable "
emsg$+= "processor in order to use this software."
ON ERROR SYS "OS_PrettyPrint", emsg$ : PRINT : END
SYS "OS_Memory", 6

ON ERROR PRINT REPORT$+" at "+STR$(ERL/10) : END

DIM code% 1024
PRINT "Assembling code"
FOR l% = 0 TO 2 STEP 2
  P% = code%
  [  OPT l%
     EXT 1                      ; Enable ExtBASICAsm

     \ Preserve our R14 and then enter SVC mode
     STMFD   R13!, {R14}
     SWI     "OS_EnterOS"

     ADR     R0, entering
     SWI     "OS_Write0"
     SWI     "OS_NewLine"

     \ Turn off all interrupts, because RISC OS won't be expecting 32 bit mode!
     SWI     "OS_IntOff"

     \ User32 code
     MOV     R0, #%10011
     MSR     CPSR_all, R0       ; Select CPSR mode, clear all the bits
     MOV     R0, R0
     
     MRS     R0, CPSR_all       ; Read CPSR
     BIC     R0, R0, #&F0000000 ; Clear the flag bits
     BIC     R0, R0, #&1F       ; Clear the mode bits
     ORR     R0, R0, #1<<30     ; Set the Z flag
     ORR     R0, R0, #%11       ; Set SVC26 mode
     MSR     CPSR_all, R0       ; Do it!
     MOV     R0, R0

     \ Turn interrupts back on again.
     SWI     "OS_IntOn"

     ADR     R0, exiting
     SWI     "OS_Write0"
     SWI     "OS_NewLine"

     \ If Z set, print Z set else print Z unset
     ADR     R0, zunset
     ADREQ   R0, zset
     SWI     "OS_Write0"
     SWI     "OS_NewLine"

     \ Return to USR mode
     BIC     R14, PC, #3
     TEQP    R14, #0
     MOV     R0, R0

     \ And get out of here.
     LDMFD   R13!, {PC}

  .zunset
     EQUS    "  Z flag is not set (this ain't right!)"
     EQUB    0
     ALIGN

  .zset
     EQUS    "  Z flag is set (as expected!)"
     EQUB    0
     ALIGN

  .entering
     EQUS    "  About to enter 32 bit mode, cross your fingers!"
     EQUB    0
     ALIGN

  .exiting
     EQUS    "  Returned to 26 bit mode. Phew!"
     EQUB    0
     ALIGN
  ]
NEXT

PRINT "Executing code"
CALL code%
PRINT "Finished"

END
Download this example
You might have got the idea that 32 bit code is not terribly useful. Under all older versions of RISC OS, this is true. In fact, as far as I'm aware, the only thing that 32 bit mode will give you on an older machine (i.e. RiscPC/IBX-100/Mico...) is: 
· Access to areas larger than 28Mb. Never really been much of an issue on RISC OS, where a web browser fits into a meg or two and serious art programs provide their own virtual memory system for those seriously huge images. 
HOWEVER, if you intend to write software that will run on the Iyonix (and we'll have Ron after you if you refuse!), you will need to take 32bit into account.
Actually, it is pretty simple. I have written a number of modules (in assembler) that run okay on an Iyonix that I don't have... How? 
· Never use MOVS PC, Rxx 
· Never use LDMFD R13!, {Rxx-Rxx, PC}^ 
· Specify your routines as corrupting processor flags, and make no attempt to preserve them 
· Don't assume anything else preserves the processor flags 
Things are a little more complicated with APCS, but since the new C compiler can make code that runs on older machines as well as the Iyonix; why not convert your projects to be 26/32 neutral? This seems like the easiest way; though all that is really required in assembler is: 
   [ {CONFIG=26}
     ; 26bit return
     MOVS    PC, R14     ; or LDMFD xxxx^
   |
     ; 32bit return
     MOV     PC, R14     ; or LDMFD xxxx
   ]
objasm will then sort out which code to use depending on whether it is assembling 26bit or 32bit... 
  
The original version of this document, and the original ARM assembler guide includes... 
· The StrongARM provides two instructions (UMUL and UMLA, IIRC) which deal with 64 bit multiplication. This is only available in 32 bit mode. 
This is incorrect. The extended multiply can be used from 26 bit mode; it's how the MP3 decoders use it! 
  
To conclude:
The change to 32bit may not sound like much, but given that all references to amending/altering the PSR bits in PC have to be changed to refer to a separate PSR not in R15, it suddenly becomes a much bigger issue. Also, you can no longer restore the PSR and branch back to the caller in one instruction, as these are now two separate operations. For that reason, code must be rewritten. 

[bookmark: swi]SWI : SoftWare Interrupt
  SWI<suffix>  <number>
This is a simple facility, but possibly the most used. Many Operating System facilities are provided by SWIs. It is impossible to imagine RISC OS without SWIs. 
Nava Whiteford explains how SWIs work (originally in Frobnicate issue 12½)... 

In this article I will attempt to delve into the working of SWIs (SoftWare Interrupts). 
What is a SWI?
SWI stands for Software Interrupt. In RISC OS SWIs are used to access Operating System routines or modules produced by a 3rd party. Many applications use modules to provide low level external access for other applications. 
Examples of SWIs are: 
· The Filer SWIs, which aid reading to and from disc, setting attributes etc. 
· The Printer Driver SWIs, used to well aid the use of the Parallel port for printing. 
· The SWIs FreeNet/Acorn TCP/IP stack SWIs used to transmit and receive data using the TCP/IP protocol usually used for sending data over the Internet. 
When used in this way, SWIs allow the Operating System to have a modular structure, meaning that the code required to create a complete operating system can be split up into a number of small parts (modules) and a module handler. 
When the SWI handler gets a request for a particular routine number it finds the position of the routine and executes it, passing any data. 
So how does it work?
Well first lets look at how you use it. A SWI instruction (in assembly language) looks like this: 
   SWI &02
or 
   SWI "OS_Write0"
Both these instructions are in fact the same, and would therefore assemble to the same instruction. The only difference is that the second instruction uses a string to represent the SWI number which is &02. When a program written using the string is used, the string is first looked up before execution. 
We're not going to deal with the strings here as they do not give a true representation of what it going on. They are often used to aid the clarity of a program, but are not the actual instructions that are executed. 
Right lets take a look at the first instruction again: 
   SWI &02
What does that mean? Well, literally it means enter the SWI handler and pass value &02. In RISC OS this means execute routine number &02. 
So how does it do that, how does it passed the SWI number and enter the SWI handler? 
If you look at a disassembly of the first 32 bytes of memory (locations 0-&1C) and disassemble them (look at the actual ARM instructions) you should see something like this: 
Address  Contents            Disassembly
00000000 : 0..å : E5000030 : STR     R0,[R0,#-48]
00000004 : .óŸå : E59FF31C : LDR     PC,&00000328
00000008 : .óŸå : E59FF31C : LDR     PC,&0000032C
0000000C : .óŸå : E59FF31C : LDR     PC,&00000330
00000010 : .óŸå : E59FF31C : LDR     PC,&00000334
00000014 : .óŸå : E59FF31C : LDR     PC,&00000338
00000018 : .óŸå : E59FF31C : LDR     PC,&0000033C
0000001C : 2¦ ã : E3A0A632 : MOV     R10,#&3200000
So what? You may think, well take a closer look. 
Excluding the first and last instructions (which are special cases) you can see that all the instruction load the PC (Program Counter), which tells the computer where to execute the next instruction from, with a new value. The value is taken from a address in memory which is also shown. (you can take a look at this for yourself using the "Read Memory" option on the !Zap main menu.) 
Now, this may seem to bare little relation to SWIs but with the following information it should make more sense. 
All a SWI does is change the Mode to Supervisor and set the PC to execute the next instruction at address &08! Putting the processor into Supervisor mode switches out 2 registers r13 and r14 and replaces these with r13_svc and r14_svc. 
When entering Supervisor mode, r14_svc will also be set to the address after the SWI instruction. 
This is really just like a Branch with Link to address &08 (BL &08) but with space for some data (the SWI number). 
As I have said address &08 contains a instruction which jumps to another address, this is the address where the real SWI Handler is! 
At this point you maybe thinking "Hang on a minute! What about the SWI number?". Well in fact the value itself is ignored by the processor. The SWI handler obtains it using the value of r14_svc that got passed. 
This is how it does it (after storing the registers r0-r12): 
1. It subtracts 4 from r14 to obtain the address of the SWI instruction. 
2. Loads the instruction into a register. 
3. Clears the last 8 bits of the instruction, getting rid of the OpCode and giving just the SWI number. 
4. Uses this value to find to address of the routine of the code to be executing (using lookup tables etc.). 
5. Restore the registers r0-r12. 
6. Takes the processor out of Supervisor mode. [actually, I don't think it does! -Rick] 
7. Jumps to the address of the routine. 
Easy! ;) 
Here is some example code, from the ARM610 datasheet: 
0x08 B Supervisor

EntryTable
 DCD ZeroRtn
 DCD ReadCRtn
 DCD WriteIRtn

 ...

Zero   EQU 0
ReadC  EQU 256
WriteI EQU 512
 
; SWI has routine required in bits 8-23 and data
; (if any) in bits 0-7.
; Assumes R13_svc points to a suitable stack

STMFD R13, {r0-r2 , R14}
 ; Save work registers and return address
LDR R0,[R14,#-4]
 ; Get SWI instruction.
BIC R0,R0, #0xFF000000
 ; Clear top 8 bits.
MOV R1, R0, LSR #8
 ; Get routine offset.
ADR R2, EntryTable
 ; Get start address of entry
 ; table.
LDR R15,[R2,R1,LSL #2]
 ; Branch to appropriate routine.

WriteIRtn
 ; Wnte with character in R0 bits 0 - 7.

.............
 LDMFD R13, {r0-r2 , R15}^
 ; Restore workspace and return, restoring
 ; processor mode and flags.
That's it, that's the basics of the SWI instruction. 
  
Sources: 
The ARM610 datasheet by Advanced Risc Machines
The ARM RISC Chip - A programmers guide by van Someren Atack published by Addison Wesley 
  
  

Thank you Nava. 
Now we'll take an example from RISC OS itself... 
1. Processor branches to &00000008 upon a SWI. 
 
2. The instruction there is a branch (old RISC OS) or an LDR into PC (newer RISC OS).
Processor is redirected... 
 
3. Preserve R10-R12. 
 
4. Mask off bits to get at the SWI call instruction address (was popped into R14). 
 
5. Get the calling SWI command. 
 
6. Mask off the condition code and SWI opcode to leave the SWI itself. 
 
7. Clear the error bit, put zero state in flags. 
 
8. No error bit set? Branch to SWI despatch. 
 
9. Set up for error, and branch to SWI despatch. 
 
10. ...etc... 
Use *MemoryI to trace through this. Here is the report for RISC OS 3.70: 

*MemoryI 8 +4
00000008 :  E59FF39C : †ô&fllig;â : LDR     PC,&000003AC
This is a 'later' version of RISC OS, so we load an address instead of branching directly to it. 

*Memory 3AC +4
Address  :     F E D C     3 2 1 0     7 6 5 4     B A 9 8 :    ASCII Data
000003AC :    01F033C0                                     : Á3ð.            
And this, is the address of our SWI despatch routine. 

*MemoryI 1F033C0
01F033C0 :  EA640146 : F.dê : B       &038038E0
01F033C4 :  E3CEC3FF : ÿãÎã : BIC     R12,R14,#&FC000003
01F033C8 :  E51CB004 : .°.â : LDR     R11,[R12,#-4]
01F033CC :  E3CBB4FF : ÿ´Ëã : BIC     R11,R11,#&FF000000 ; =&FF<<24
01F033D0 :  E92D0800 : ..-é : STMFD   R13!,{R11}
01F033D4 :  E3DBB802 : .¸Ûã : BICS    R11,R11,#&00020000 ; =1<<17
01F033D8 :  0A6401A6 : ¦.d. : BEQ     &03803A78
01F033DC :  E38EA003 : . ‰ã : ORR     R10,R14,#3
01F033E0 :  E33AF000 : .ð:ã : TEQP    R10,#0
01F033E4 :  E35B0017 : ..[ã : CMP     R11,#&17           ; =23 ; *** Not R8-R14
01F033E8 :  135B0034 : 4.[. : CMPNE   R11,#&34           ; ="4" (52)
01F033EC :  13CEE201 : .âÎ. : BICNE   R14,R14,#&10000000 ; =1<<28
01F033F0 :  E35B0C01 : ..[ã : CMP     R11,#&0100         ; =256
01F033F4 :  379FF10B : .ñ.7 : LDRCC   PC,[PC,R11,LSL #2]
01F033F8 :  EA6401B8 : ¸.dê : B       &03803AE0
[...non-SWI stuff snipped...]
Part of the SWI despatch. 

*MemoryI 38038E0
038038E0 :  E92D1C00 : ..-é : STMFD   R13!,{R10-R12}
038038E4 :  E14FB000 : .°Oá : MRS     R11,SPSR           ; ARMv3 and later
038038E8 :  E20BC23F : ?â.â : AND     R12,R11,#&F0000003
038038EC :  E18EE00C : .à‰á : ORR     R14,R14,R12
038038F0 :  E20BC0C0 : ÀÀ.â : AND     R12,R11,#&C0       ; ="À" (192)
038038F4 :  E18EEA0C : .ê‰á : ORR     R14,R14,R12,LSL #20
038038F8 :  E10FC000 : .À.á : MRS     R12,CPSR           ; ARMv3 and later
038038FC :  E3CCC01F : .ÀÌã : BIC     R12,R12,#&1F       ; =31
03803900 :  E38CC003 : .À…ã : ORR     R12,R12,#3
03803904 :  E129F00C : .ð)á : MSR     CPSR_c_f,R12       ; ARMv3 and later
03803908 :  EA9BFEAD : -Þ.ê : B       &01F033C4
[...non-SWI stuff snipped...]
The beginning of the SWI despatch calls the code above. This is used on RiscPC-alikes because SWIs occur in SVC32 so the code fudges it to be SVC26 so RISC OS will work.
This doesn't occur on ARM2 or ARM3 machines, and the Iyonix runs in 32bit modes so, again, it won't occur. 
Finally, there are two calls into this chunk of code... 

*MemoryI 3803A78
03803A78 :  E33FF003 : .ð?ã : TEQP    PC,#3
03803A7C :  E3CEE201 : .âÎã : BIC     R14,R14,#&10000000 ; =1<<28 ; *** Not R8-R14
03803A80 :  E92D4000 : .@-é : STMFD   R13!,{R14}
03803A84 :  E59BB7E4 : ä·.â : LDR     R11,[R11,#&7E4]    ; =2020
03803A88 :  E35B050E : ..[ã : CMP     R11,#&03800000     ; =7<<23
03803A8C :  3A000004 : ...: : BCC     &03803AA4
03803A90 :  292D8000 : ..-) : STMCSFD R13!,{PC}          ; *** Offset not guaranteed
03803A94 :  2A009CDD : Ý†.* : BCS     &0382AE10
03803A98 :  E1A00000 : .. á : NOP
03803A9C :  E8BD4000 : .@½è : LDMFD   R13!,{R14}
03803AA0 :  EA9BFE39 : 9Þ.ê : B       &01F0338C
03803AA4 :  E3A0A003 : .  ã : MOV     R10,#3
03803AA8 :  EB0016FD : Ý..ë : BL      &038096A4
03803AAC :  EAFFFFFA : úÿÿê : B       &03803A9C
03803AB0 :  E1A0B00E : .° á : MOV     R11,R14
03803AB4 :  EF02010A : ...ï : VDUX    &0A                ; =10
03803AB8 :  7F02010D : .... : VDUVCX  &0D                ; =13
03803ABC :  E1A0E00B : .À á : MOV     R14,R11
03803AC0 :  EA9BFE31 : 1Þ.ê : B       &01F0338C
03803AC4 :  E1A0A000 : .  á : MOV     R10,R0
03803AC8 :  E20B00FF : ÿ..â : AND     R0,R11,#&FF        ; ="ÿ" (255)
03803ACC :  E1A0B00E : .° á : MOV     R11,R14
03803AD0 :  EF020000 : ...ï : SWI     "XOS_WriteC"
03803AD4 :  71A0000A : .. q : MOVVC   R0,R10
*MemoryI 3803AD8
03803AD8 :  E1A0E00B : .À á : MOV     R14,R11
03803ADC :  EA9BFE2A : *Þ.ê : B       &01F0338C
03803AE0 :  E35B0C02 : ..[ã : CMP     R11,#&0200         ; =512
03803AE4 :  3AFFFFF6 : öÿÿ: : BCC     &03803AC4
03803AE8 :  E92D4200 : .B-é : STMFD   R13!,{R9,R14}
03803AEC :  E20EA33F : ?£.â : AND     R10,R14,#&FC000000 ; =&3F<<26
03803AF0 :  E28FE04B : KÀ·â : ADR     R14,&03803B43
03803AF4 :  E18EE00A : .À‰á : ORR     R14,R14,R10
03803AF8 :  E3CBC03F : ?Àëã : BIC     R12,R11,#&3F       ; ="?" (63)
03803AFC :  E1A0A22C : ,¢ á : MOV     R10,R12,LSR #4
03803B00 :  E20AA03F : ? .â : AND     R10,R10,#&3F       ; ="?" (63)
03803B04 :  E59AA790 : .§.â : LDR     R10,[R10,#&790]    ; =1936
03803B08 :  E35A0000 : ..Zã : CMP     R10,#0
03803B0C :  0A000009 : .... : BEQ     &03803B38
03803B10 :  E59A900C : .'.â : LDR     R9,[R10,#&00C]     ; =12
03803B14 :  E159000C : ..Yá : CMP     R9,R12
03803B18 :  159AA008 : . .. : LDRNE   R10,[R10,#8]
03803B1C :  1AFFFFF9 : ùÿÿ. : BNE     &03803B08
03803B20 :  E89A1400 : ...è : LDMIA   R10,{R10,R12}
03803B24 :  E59CC00C : .À†â : LDR     R12,[R12,#&00C]    ; =12
03803B28 :  E35C0000 : ..\ã : CMP     R12,#0
03803B2C :  120BB03F : ?°.. : ANDNE   R11,R11,#&3F       ; ="?" (63)
03803B30 :  128CC004 : .À…. : ADDNE   R12,R12,#4
03803B34 :  11A0F00A : .ð . : MOVNE   PC,R10
I'll be nice and leave you to work out what that code does... 

[bookmark: cmn]CMN : Compare Negative
  CMN<suffix>  <op 1>, <op 2>

                status = op_1 - (- op_2)
CMN is the same as CMP, except it allows you to compare against small negative values (the logical NOT of operand two) that would be hard to implement otherwise; such as -1 to end a list.
So to compare with -1 we would use: 
  CMN     R0, #1                  ; Compare R0 with -1
Refer also to the details for the CMP instruction. 
  
  
[bookmark: cmp]CMP : Compare
  CMP<suffix>  <op 1>, <op 2>

                status = op_1 - op_2
CMP allows you to compare the contents of a register with another register or an immediate value, updating the status flags to allow conditional execution to take place.
It performs a subtraction, but does not store the result anywhere. Instead, the flags are updated as appropriate.
The flags refer to operand one compared against operand two. Thus, the GT suffix will be executed if operand one is greater than operand two.
Obviously you do not need to explicitly specify the S suffix as the point of the command is to update the status flags... If you do specify it, it will be ignored. 
  
  
[bookmark: teq]TEQ : Test Equivalence
  TEQ<suffix>  <op 1>, <op 2>

                Status = op_1 EOR op_2
TEQ is similar to TST. The difference is that the notional arithmetical calculation is an EOR rather than an AND.
This provides a way to see if bits in both operands are the same or not without affecting the Carry flag (unlike CMP).
TEQ is also used with the P suffix to alter the flags in R15 (in 26-bit mode). Refer to psr.html for more details, or go here for how to do it in 32-bit mode. 
  
  
[bookmark: tst]TST : Test bits
  TST<suffix>  <op 1>, <op 2>

                Status = op_1 AND op_2
TST, like CMP, does not produce a result to be placed into a destination register. Instead it performs an operation on the two operands given and reflects the result of this in the status flags.
TST is used to see if a particular bit is set. Operand one is the data word to test and operand two is a bit mask. After testing, the Zero flag will be set upon a match, or otherwise clear.
Like CMP, it does not matter if you specify the S suffix or not. 
  TST     R0, #%1                 ; Test if bit zero is set in R0


	[image: http://www.heyrick.co.uk/assembler/arm3.gif]
	Registers
and
Processor Modes
	[image: http://www.heyrick.co.uk/assembler/arm3.gif]


  
The ARM processor has twenty seven registers, some of which have conditions applied, so you only get to use sixteen at any one time... 
· Register 0 to register 7 are general purpose registers and can be used for ANY purpose.
Unlike 80x86 processors which require certain registers to be used for stack access, or the 6502 which places the result of mathematical calculations in the Accumulator, the ARM processor is highly flexible in its register use. 
· Register 8 to register 12 are general purpose registers, but they have shadow registers which come into use when you switch to FIQ mode. 
· Register 13 is typically the OS stack pointer, but can be used as a general purpose register. This is an operating system issue, not a processor issue, so if you don't use the stack you can corrupt this freely within your own code as long as you restore it afterwards. Each of the processor modes shadow this register. 
· Register 14 is dedicated to holding the address of the return point to make writing subroutines easier. When you branch with link (BL) the return address is stored in R14. 
Likewise when the program is first run, the exit address is stored in R14. All instances of R14 must be preserved in other registers (not really efficient) or in a stack. This register is shadowed across all of the processor modes. This register can be used as a general purpose register once the link address has been preserved. 
· Register 15 is the program counter. It holds the status of the processor as well as a twenty-six bit number which is the address of the program currently being used. 
  
  
To make this a little clearer... Another chart: 

User Mode  SVC Mode   IRQ Mode   FIQ Mode  APCS

R0 ------- R0 ------- R0 ------- R0        a1
R1 ------- R1 ------- R1 ------- R1        a2
R2 ------- R2 ------- R2 ------- R2        a3
R3 ------- R3 ------- R3 ------- R3        a4
R4 ------- R4 ------- R4 ------- R4        v1
R5 ------- R5 ------- R5 ------- R5        v2
R6 ------- R6 ------- R6 ------- R6        v3
R7 ------- R7 ------- R7 ------- R7        v4
R8 ------- R8 ------- R8         R8_fiq    v5
R9 ------- R9 ------- R9         R9_fiq    v6
R10 ------ R10 ------ R10        R10_fiq   sl
R11 ------ R11 ------ R11        R11_fiq   fp
R12 ------ R12 ------ R12        R12_fiq   ip
R13        R13_svc    R13_irq    R13_fiq   sp
R14        R14_svc    R14_irq    R14_fiq   lr
------------- R15 / PC -------------       pc
The rightmost column is a list of names used for APCS code, refer here for details of APCS assembler. 
  
  
The Program Counter is built up as follows: 
  Bit  31  30  29  28  27  26  25------------2  1  0

       N   Z   C   V   I   F   Program Counter  S1 S0
For further explanation of R15, refer to psr.html. 
  
  
By now you may be wondering about these "modes", such as "FIQ" mentioned above. 
· User Mode, the usual mode for applications to run in. Your memory access is restricted and you cannot read directly from hardware devices. 
· Supervisor Mode (SVC Mode), used mainly by SWIs and the OS. This mode has additional privileges which allow greater control of the computer. For example, you have to go to Supervisor Mode in order to read from a podule. It cannot be done in User Mode. 
· Interrupt Mode (IRQ Mode), used to handle peripherals that issues interrupts. This mode is also privileged. Such devices causing IRQs are the keyboard, the VSync (when the screen refresh is occurring), IOC timers, serial, harddisc, floppy etc etc... 
· Fast Interrupt Mode (FIQ Mode), used to handle peripherals that issue fast interrupts. This mode is also privileged. Such devices causing FIQs are the floppy disc handling data, the serial port (on 82C71x machines such as the A5000) and Econet. 
The difference between IRQ and FIQ is with FIQ you have to process your stuff as quickly as possible and then get the .... out of there. An IRQ may be interrupted by an FIQ but an IRQ cannot interrupt an FIQ. To make FIQs faster, they have more shadow registers. FIQs cannot call SWIs. FIQs must also disable interrupts. If it becomes necessary for an FIQ routine to re-enable interrupts, it's too slow and should be IRQ not FIQ. Phew! 

